
Введение 
 

В настоящее время наиболее распространенными операционными сис-
темами для персональных компьютеров являются системы семейства 
Windows компании Microsoft. Для современного программиста совершенно 
необходимо овладеть навыками разработки программного обеспечения для 
достаточно сложной для программирования операционной системы. Одной 
из особенностей программирования для Windows является то, что при напи-
сании программ важная роль уделяется созданию и кодированию пользова-
тельского интерфейса. 

Для повышения производительности программистов и улучшения качества 
получаемых Windows-программ разработчиками программного обеспечения 
было предпринято несколько шагов. Первый шаг – отход от традиционного 
процедурного и модульного программирования в пользу объектно-
ориентированного программирования и создание специализированных объект-
но-ориентированных библиотек. Второй шаг – использование интегрированных 
сред программирования, сочетающих редактор, компилятор, отладчик и объ-
ектно-ориентированную библиотеку. И, наконец, третий шаг – появление ин-
тегрированных сред, в которых разработка пользовательского интерфейса 
представляет собой визуальный процесс, автоматически сопровождаемый гене-
рацией соответствующего программного кода. Подобные системы получили 
название RAD-сред (Rapid Application Development – быстрая разработка при-
ложений). Delphi представляет собой одну из таких систем, выгодно сочетая 
простоту освоения и программную мощь. 

Систему Delphi составляют: 
1. Интегрированная среда разработки (IDE), включающая редактор кода, 

визуальный редактор интерфейса приложения, компилятор и отладчик. 
2. Объектная библиотека VCL (в Delphi 6 добавлена библиотека CLX). 
3. Компилятор командной строки. 
4. Средства для работы с базами данных. 
5. Файлы справки. 
6. Файлы исходного кода объектной библиотеки VCL. 
Система Delphi поставляется в трех вариантах: 
а) основная версия (издание «Standard») рассчитана на новичков в Delphi и 

непрофессионалов; 
б) второй уровень (издание «Professional») предназначен для профессио-

нальных разработчиков и включает дополнительно к изданию «Standard» под-
держку баз данных, расширенную поддержку программирования для Internet и 
некоторые дополнительные инструменты; 

в) полное издание («Enterprise») ориентировано на разработчиков прило-
жений масштаба предприятия. 

1. Понятия «проект», «форма», «компонент» 
 

При написании программ начального уровня для MS-DOS используется, 
как правило, один файл с исходным текстом программы. Для разработки даже 
простейших приложений Windows, как правило, необходимо несколько исход-
ных файлов. Эти файлы логически объединены понятием проекта. Проект 
(Project) – набор файлов, необходимых для создания приложения. 

Большинство приложений Windows выполняются в отдельных областях 
экрана, называемых окнами. Кроме того, приложения часто создают дополни-
тельные окна в процессе работы. Специфика создания Windows-приложения в 
любой RAD-среде, в том числе и в Delphi, заключается в разбиении исходной 
задачи на две подзадачи: разработка интерфейсной части (как программа вы-
глядит и как взаимодействует с пользователем) и разработка программной час-
ти (что программа делает). Окно на этапе разработки в IDE называется формой 
(Form). 

Несмотря на разнообразие внешнего вида окон приложений, в них можно 
выделить некоторые стандартные элементы, характерные практически для лю-
бого окна (например, меню, кнопки, строка состояния, текстовые поля). На эта-
пе разработки этим элементам соответствуют так называемые компоненты 
(Components). Компоненты подобны кирпичикам, из которых строится интер-
фейс приложения. Отметим, что кроме компонентов, предназначенных для по-
строения интерфейса (так называемых визуальных компонентов), система 
Delphi предлагает большой набор невизуальных компонентов, которые на этапе 
выполнения приложения не видны, однако могут использоваться приложением 
для организации доступа к данным, отсчета временных промежутков, связи с 
Internet. Итак, компонент является частью формы, форма – это часть проекта. 
На этапе выполнения проекту соответствует приложение, форме – окно, а ком-
поненту – один из стандартных элементов окна. 

Рассмотрим подробнее состав проекта. Типичный проект, как правило, 
включает следующие файлы: 

1) главный – текстовый файл с расширением DPR, содержащий главный 
программный блок; главный файл проекта подключает все остальные исполь-
зуемые файлы и содержит операторы для запуска приложения; 

2) описания форм – текстовые файлы с расширением DFM, которые содер-
жат описание внешнего вида формы и размещенных на ней компонентов, уста-
новленного на этапе проектирования; количество файлов описания форм равно 
количеству форм в проекте; 

3) программные модули – текстовые файлы с расширением PAS, которые 
содержат исходный код на языке Object Pascal. Это может быть либо код про-
граммного описания формы, либо участок программы, выделенный в отдель-
ный модуль; количество файлов программных модулей больше или равно ко-
личеству форм в проекте; 

4) файлы ресурсов (в частности, файл ресурсов проекта) – двоичные фай-
лы с расширением RES, содержат ресурсы Windows; файл ресурсов проекта 
включает пиктограмму приложения; 

5) файл опций проекта – текстовый файл с расширением DOF, содержит на-
стройки компилятора и проекта и используется, если эти настройки отличны от 
принятых по умолчанию. 

Проект также может содержать графические файлы (расширение BMP, JPG), 
файлы оперативной подсказки (расширение HLP) и другие файлы. 

Файлы программных модулей доступны для редактирования в IDE. Код 
программного описания форм создается Delphi автоматически, обычно редак-
тируются в нем лишь участки, ответственные за обработку событий компонент. 
Главный файл проекта также генерируется автоматически. Необходимость ре-
дактирования его возникает крайне редко. Для просмотра этого файла можно 
использовать команду IDE Project|View Source . Изменить пиктограмму 
приложения возможно при помощи команды IDE Project| Options| 

Application| Icon. 

2. Сравнение языков Object Pascal и Turbo Pascal 
 
Язык Object Pascal является встроенным языком программирования систе-

мы Delphi и развитием языка Turbo Pascal, используя некоторые особенности 
диалекта этого языка. Поэтому Object Pascal будет рассмотрен в сравнении с 
версией Turbo Pascal 7.0. 

Заметим, что для использования некоторых специфических особенностей 
Object Pascal предусмотрен так называемый режим расширенного синтаксиса. 
Для управления режимом используется директива компилятора $X (по умолча-
нию режим включен – $X+). 

Алфавит Object Pascal совпадает с алфавитом Turbo Pascal. Остались 
прежними правила записи констант и идентификаторов. Набор ключевых слов 
и директив языка существенно расширен в сравнении с Turbo Pascal. Однако, 
так как в редакторе IDE все ключевые слова и стандартные директивы выделя-
ются полужирным шрифтом, то запоминать их для того, чтобы не использо-
вать в качестве идентификаторов программиста, нет необходимости. 

В Object Pascal появилась новая форма записи комментариев. Теперь ком-
ментарием является также любой текст от двойной наклонной черты до конца 
строки. Старые способы записи комментариев сохранены. Новый способ не 
применим для записи директив компилятора: 

A+B; // это комментарий 
 

2.1. Типы данных 
 

Набор типов данных претерпел изменения, связанные с переходом к 32-
разрядной платформе программирования. Он также был дополнен некоторыми 
новыми типами, отсутствующими в Turbo Pascal. 

Типы Byte, Word, ShortInt, LongInt по диапазону и формату соответст-
вуют одноименным типам Turbo Pascal. Часто используемый тип Integer име-
ет формат 4 байта и представляет диапазон от -2147483648 до 2147483647. 
Появились новые целые типы: SmallInt (формат – 2 байта, диапазон от –32768 
до 32767) и Cardinal (формат – 4 байта, диапазон 0...4294967295). В Delphi 4 
дополнительно введены типы LongWord (является синонимом Cardinal) и 
Int64 (64-разрядное знаковое целое, до 18 значащих цифр). 

Вещественные типы Single, Double, Extended, Comp аналогичны типам 
Turbo Pascal. Заметим, что операции с типами Single, Double, Extended и 
Comp поддерживаются аппаратно, на уровне микропроцессора. Учитывая попу-
лярность идентификатора Real среди Pascal-программистов, в Delphi 4 этот тип 
объявлен как синоним типа Double. Для обратной совместимости в Delphi 4 
описан тип Real48, аналогичный по формату типу Real ранних версий Delphi и 
Turbo Pascal. Новый вещественный тип Currency рекомендован как тип для 
финансовых расчетов. Его формат – 8 байт, но он имеет всего четыре значащих 
цифры после запятой. Отметим, что внутренним представлением вещественных 

констант в Object Pascal является тип Extended. Это может служить источни-
ком трудноуловимых ошибок, аналогичных приведенному примеру: 

var A: Double; 

... 

A := 1.1; 

if A = 1.1 then ...// в данном примере условие будет ложно! 

Упражнение. Разберите, почему оператор if работает подобным образом. 
Почему оператор работает корректно, если присвоить переменной A значение 
1.5 и сравнить его с константой 1.5 (сравните двоичные представления чисел 
1.1 и 1.5)? 

Для представления отдельных символов Object Pascal использует типы 
AnsiChar, WideChar и Char. Типы Char и AnsiChar являются синонимами. Они 
кодируют один из 256 символов таблицы ANSI, их формат – один байт. Тип 
WideChar занимает два байта и кодирует символы из кодовой таблицы Unicode. 

Набор булевых типов данных был расширен для большей совместимости с 
программными библиотеками операционной системы. Кроме стандартного ти-
па Boolean появились три новых типа: ByteBool, WordBool, LongBool. Их 
формат – соответственно один, два и четыре байта, нулевое значение интерпре-
тируется как «ложь», любое другое – как «истина». 

Для представления дат и времени в Object Pascal используется тип 
TDateTime. Фактически это синоним типа Double, т.е. вещественное число. Це-
лая часть этого числа соответствует количеству дней, прошедших с 30 декабря 
1899 года, дробная – части суток, прошедших с полуночи. 

Object Pascal разделяет множество строковых типов на три вида: короткие 
строки, длинные и строки с завершающим нулем. Короткие строки объявля-
ются при помощи ключевого слова string, за которым следует заключенное в 
квадратные скобки значение максимальной длины строки. Формат такого типа 
полностью совпадает с форматом аналогичного объявления в Turbo Pascal. Для 
объявления короткой строки длиной 255 символов возможно использование 
синонима ShortString. 

Длинные строки объявляются при помощи ключевого слова string без 
указания максимальной длины строки. Максимальный размер длинной строки 
теоретически может составлять около 2 Гб. В переменной, объявленной как 
длинная строка, фактически хранится адрес начала строки в памяти. Символы 
длинной строки индексируются от 1 до N+1, где N – реальная длина строки. 
Доступ к отдельным символам организуется так же, как и для короткой строки, 
например St[2], St[2345], при этом символ с индексом N+1 всегда равен #0. 
Для получения значения длины строки следует использовать функцию Length, 
для изменения длины – процедуру SetLength. Для длинной строки в памяти 
хранятся количество символов в строке и счетчик ссылок на строку, которые 
являются 32-разрядными беззнаковыми целыми числами. Счетчик ссылок на 
строку позволяет экономно расходовать память при наличии разных перемен-
ных, содержащих одинаковые строки. Синонимом типа string является тип 

AnsiString. Тип данных WideString – это длинная строка, состоящая из сим-
волов таблицы Unicode (код символа занимает два байта). Тип WideString не 
поддерживает счетчик ссылок на строку. 

Для объявления строки с завершающим нулем (так называемый ANSIIZ 
формат) используется идентификатор PChar. Переменная типа PChar – это ука-
затель на массив Char, который индексируется с нуля и заканчивается симво-
лом #0. Строки с завершающим нулем используются для совместимости с про-
граммами, написанными на языках C и C++. В режиме расширенного синтакси-
са значения указателей PChar можно складывать и вычитать, массив символов с 
нижним индексом, равным нулю, и строковые константы совместимы с типом 
PChar. 

Object Pascal поддерживает автоматическое приведение типов для длин-
ных и коротких строк. Явное преобразование строк выполняется с помощью 
конструкций вида ShortString(S) и AnsiString(S). 

В Object Pascal включена поддержка переменных вариантного типа. Та-
кие переменные могут принимать значения различных типов – быть целым или 
вещественным числом, строкой, массивом, принимать значения булевых пере-
менных. Идентификатором вариантного типа служит ключевое слово Variant. 
Для вариантных переменных поддерживается гибкое встроенное приведение 
типов. Следующий пример показывает это: 

var V1, V2, V3, V4: Variant; //объявление переменных 

. . . 

V1 := 5; //целое число 

V2 := 0.8; //вещественное число 

V3 := '10.2'; //строка 

V4 := V1+V2+V3; //целое число, равно 16 

Будьте внимательны с автоматическим приведением типов. Обратимся к 
примеру: 

var V1, V2, V3, V4: Variant; //объявление переменных 

. . . 

V1 := '100'; //строка 

V2 := '10'; //строка 

V3 := 1; //число 

V4 := V1 + V2 + V3; 

Чему будет равно V4? Эта переменная будет содержать число, и так как 
операция сложения левоассоциативна, то число это будет равно 10011. 

Внутреннему представлению вариантной переменной соответствует струк-
тура VVarData. Ее подробное описание можно найти в интерактивной системе 
помощи. Отметим, что одно из полей этой структуры позволяет узнать текущий 
тип вариантной переменной. 

Object Pascal содержит большое количество подпрограмм для работы с ва-
риантными переменными. В качестве примера рассмотрим функцию VarArray-
Create, создающую массив произвольной размерности и произвольного типа и 
помещающую его в вариантную переменную: 

var V: variant; 

. . . 

V := VarArrayCreate([1,10,5,7], varDouble); 

Первый параметр функции VarArrayCreate – открытый массив целых, ко-
торый задает размерность массива и границы каждой размерности (в нашем 
примере: две размерности, границы первой – от 1 до 10, второй – от 5 до 7). 
Второй параметр – тип значения элементов массива (в нашем примере – веще-
ственные числа). 

Операции с вариантными переменными выполняются сравнительно мед-
ленно, их присутствие в программе может понадобиться только для поддержки 
некоторых программных технологий (COM, ActiveX). 

В Delphi 4 появилась возможность работать с динамическими массивами. 
Работу с такими массивами иллюстрирует следующий пример: 

var A: array of Integer; {динамический массив объявляется без 

указания диапазона индексов} 

. . . 

  SetLength(A,10); {перед использованием надо задать длину мас-

сива в логических элементах при помощи процедуры SetLength} 

  A[0] := 10; //индексы начинаются с нуля 

  A[9] := -100; 

  A := nil; //память, занимаемая массивом, освобождается 

Отметим, что после первого использования процедуры SetLength все эле-
менты созданного массива обнуляются. Если вызвать SetLength для уже соз-
данного массива с целью увеличения его размера (SetLength(A,20)), то старые 
значения элементов в массиве сохранятся. 

Имеется возможность создавать многомерные динамические массивы: 
var B: array of array of integer; 

. . . 

  SetLength(B, 2); 

  SetLength(B[0], 3); 

  SetLength(B[1], 2); 

  B[0,0] := 10; 

  B[1,1] := -100; 

Для массивов можно использовать следующие функции: Low(A) и High(A) – 
возвращают значения нижней и верхней границы индекса массива A, 
Length(A) – возвращает количество элементов массива. Для динамических 

массивов функция Low всегда возвращает значение 0. Если для динамического 
массива не вызывалась процедура SetLength, то значение функции High будет 
равно -1. Функция Copy(A, Start, Count) возвращает динамический массив, 
являющийся частью динамического массива A. Так как переменные динамиче-
ских массивов фактически являются указателями, то присваивание вида A := B 
(A, B – динамические массивы одного типа) приведет к тому, что обе перемен-
ные будут ссылаться на один участок памяти, и память для A не будет выделе-
на. Внутри функции Copy создается динамический массив и возвращается 
ссылка на него, так что после оператора A := Copy(B, 0, 10) длина массива A 
будет равна 10, даже если она предварительно была установлена равной 100. 

 
2.2. Переменные и константы 

 
Типизированные константы при установке директивы компилятора $J+ 

становятся константами настоящими, т.е. их значение нельзя изменять. Начи-
ная с Delphi 3, описание строковых констант можно размещать в секции 
resourcestring. Строковые константы, помещенные в этой секции, при ис-
пользовании не отличаются от обычных, однако находятся в специальной части 
исполняемого файла, называемой ресурсной строковой таблицей. Это позволя-
ет изменять такие константы в откомпилированных файлах при помощи специ-
альных программ-редакторов ресурсов. 

Все глобальные переменные в Object Pascal инициализируются нулевым 
значением (целые и вещественные числа получают значение 0, логические пе-
ременные – false, указатели – nil, строки инициализируются пустой строкой). 
Существует возможность задать начальное значение глобальной переменной 
при ее объявлении: 

var X: Integer = 10; 

Подчеркнем, что техника инициализации работает только для глобальных 
переменных, т.е. объявленных в основной программе или в модуле, но не в 
подпрограмме. 

 
2.3. Процедуры и функции 

 
В Object Pascal функции могут возвращать значения любого типа, за ис-

ключением файлового. При использовании расширенного синтаксиса возможен 
вызов функции как процедуры, т.е. без присвоения возвращаемого функцией 
значения. 

При описании подпрограмм возможно задание специальных директив, 
регулирующих порядок передачи параметров. Директива register указывает 
на передачу параметров по возможности через регистры процессора. Этот 
способ является быстрым и используется в Object Pascal по умолчанию. Ди-
ректива pascal задает традиционный способ передачи параметров через стек. 
Директива   cdecl   применяется  при   описании  процедур,  которые  должны 

быть совместимы с языком C. Для работы с системными библиотеками описы-
ваются процедуры с директивой stdcall. Директива safecall уведомляет 
компилятор о необходимости использования безопасного метода передачи па-
раметров процедуре с дополнительным контролем. При использовании дирек-
тив register и pascal параметры передаются, начиная с самого левого пара-
метра. При использовании остальных директив передача параметров начинает-
ся с правого параметра: 

function F(X: Integer): Byte; stdcall; //пример директивы 

При использовании расширенного синтаксиса в функциях можно исполь-
зовать внутреннюю переменную Result. Тип этой переменной совпадает с ти-
пом возвращаемого значения функции, описывать ее в блоке var функции не 
надо. Эта переменная может находиться как слева, так и справа от знака при-
сваивания. Для задания функции возвращаемого значения достаточно присво-
ить это значение переменной Result: 

function Sum(A, B: Integer): Integer; 

begin 

  Result := A + B; //эквивалентно Sum := A + B 

end; 

В качестве параметров в подпрограммы можно передавать так называемые 
открытые массивы, т.е. массивы, размер которых заранее неизвестен и уточ-
няется непосредственно при вызове подпрограммы: 

function Sum(const A: array of Integer): Integer; 

var i: Integer; 

begin 

  Result := 0; //переменную Result необходимо инициализировать 

  for i := Low(A) to High(A) do //для открытого массива Low(A)=0 

    Result := Result + A[i] 

end; 
В приведенном примере в функцию Sum можно передать любой массив с 

элементами Integer или даже сконструировать такой массив непосредственно 
во время передачи: 

k := Sum([0, 5, 6, X*3]); 

В качестве фактического параметра вместо массива можно передать един-
ственное значение с типом открытого массива. В подпрограмме это значение 
трактуется как массив из одного элемента: 

k := Sum(5); 

Особый случай представляет объявление array of const. При наличии 
такого объявления в процедуру можно передать открытый массив из элементов 
разных типов. Каждый элемент такого массива преобразуется в запись типа 

TVarRec. У этой записи есть поле VType: Byte, при помощи которого можно 
узнать конкретный тип элемента: 

function Example(const A: array of const): Integer; 

. . . 

  case A[i].VType of 

    vtInteger:. . . 

    vtExtended:. . . 

В Delphi 4 появилась возможность объявлять подпрограммы с одинаковы-
ми именами (перегружать подпрограммы). Для этого необходимо указать ди-
рективу overload после описания параметров процедуры или функции. Чтобы 
компилятор смог различить вызовы таких подпрограмм, у них должны отли-
чаться либо количество, либо тип параметров: 

procedure Ov1(X: Double); overload; 

. . . 

procedure Ov1(X: Integer); overload; 

Если в программе появится вызов Ov1(1.6), это будет означать вызов пер-
вой процедуры, вызов Ov1(4) будет означать вызов второй процедуры. 

Начиная с Delphi 4, стало возможным описывать в объявлении подпро-
грамм параметры по умолчанию: 

procedure Proc(X: Integer; Y: Real = 3.14); 

Для приведенного выше объявления возможны вызовы Proc(7) и 
Proc(7, 4.6). В первом случае X=7, Y=3.14, во втором случае X=7, Y=4.6. Ес-
ли объявление содержит параметры по умолчанию, то они должны быть по-
следними в списке параметров. 

 
2.4. Модули 

 
В Object Pascal модуль может содержать секции initialization и 

finalization. Секцию initialization составляют операторы, которые вы-
полняются при подключении модуля. Секция finalization, которая может 
присутствовать только при наличии секции initialization, включает опера-
торы, выполняемые при завершении программы при выгрузке модуля. 

В составе Delphi имеется большое количество модулей. Основная часть их 
составляет объектную библиотеку VCL. Модуль Math содержит несколько де-
сятков процедур и функций для математических и статистических расчетов. 
Модуль SysUtils включает процедуры и функции для работы с файлами, стро-
ками, переменными типа TDateTime, конвертирования типов и многие другие. 
Этот модуль подключается к большинству проектов автоматически. 

Рассмотрим некоторые процедуры и функции модуля SysUtils. Процеду-
ра Beep генерирует звуковой сигнал. Функция Now возвращает текущую дату и 
время в формате TDateTime. Функции Date и Time возвращают соответственно 

текущую дату и время. Функции DateTimeToStr, DateToStr, TimeToStr кон-
вертируют значения переменной типа TDateTime в строку. Обратное преобра-
зование выполняют функции StrToDateTime, StrToDate, StrToTime. Для по-
лучения номера дня недели по дате служит функция DayOfWeek, при этом зна-
чение 1 соответствует воскресенью. Получить величину в формате TDateTime 
по значению года, месяца и дня позволяет функция EncodeDate. EncodeTime 
кодирует время по переданным значениям часов, минут, секунд и миллисекунд. 
Обратные преобразования выполняют функции DecodeDate и DecodeTime. 
Функция IsLeapYear возвращает значение «истина», если переданное ей целое 
значение соответствует високосному году. 

Функции IntToStr, FloatToStr, FloatToStrF, CurrToStr, StrToInt, 
StrToFloat используются для перевода числовых значений в строковые и на-
оборот. 

Функция Format используется для так называемого форматного вывода 
(подобного применяемому в языке C). Аргументы функции – строка, содержа-
щая спецификаторы формата, и открытый массив значений. Функция возвра-
щает переданную строку, в которой спецификаторы формата заменены значе-
ниями из массива: 

S:=Format('Задано %d и %d', [5,2]); //S='Задано 5 и 2' 

Спецификаторы формата имеют вид 

 %[<индекс>:][-][<ширина>][.<точность>]<тип>, 
где <индекс> – номер аргумента в массиве, к которому относится специфика-
тор; [-] – идентификатор выравнивания влево; <ширина> – ширина поля; <точ-
ность> – спецификатор точности; <тип> – обязательное поле, ниже перечисле-
ны его возможные значения: 

d – десятичное число; 
e – формат вида -d.dd..e+ddd; если задана <точность>, то она обозначает 

общее количество цифр; 
f – формат вида -d.ddd; <точность> – количество знаков после запятой; 
g – совпадает с e или f, в зависимости от того, что дает меньше цифр; 
n – совпадает с f, дополнительно содержит разделители тысяч; 
m – монетарный формат (добавляет $ или р.); 
p – формат для указателей (XXXX:YYYY); 
s – строка, если есть <точность>, то это количество символов строки; 
x – шестнадцатеричный формат целых чисел. Би

3. Объектно-ориентированное программирование. 
Базовые понятия 

 
Традиционное процедурное программирование подразумевает разработку 

программы методом последовательного уточнения и детализации. Исходная за-
дача разбивается на несколько связанных подзадач, затем каждая подзадача де-
лится на более мелкие, и так далее, вплоть до неких элементарных задач. Объ-
ектно-ориентированное программирование (ООП) предполагает представление 
исходной задачи как процесса взаимодействия некоторых вполне самостоя-
тельных программных единиц, называемых объектами. Так, в среде Delphi при 
разработке приложения представляет объект, любой компонент также является 
объектом. Работа программиста в этом случае заключается в наиболее полном 
описании набора объектов задачи и кодировании их связей. Подобный стиль 
программирования позволяет существенно сократить время разработки слож-
ных программ и программных комплексов, особенно при использовании гото-
вых наборов объектов, так называемых объектных библиотек. 

Наиболее распространенными языками, реализующими ООП, являются 
C++, Java, Turbo Pascal (начиная с версии 5.5), Object Pascal. При программиро-
вании на этих языках используются такие объектные библиотеки, как Turbo 
Vision, OWL, MFC, VCL и другие. 

ООП основывается на трех принципах: 
1. Инкапсуляция – объединение в одном программном типе, называемом 

класс, как данных, так и подпрограмм для их обработки. Данные класса хранятся в 
полях класса, подпрограммы для их обработки называются методами класса. 

2. Наследование – возможность создавать новые классы на основе сущест-
вующих. Новый класс называется классом-потомком (или дочерним классом, 
производным классом), старый – классом-предком (или родительским классом, 
базовым классом). Наследование предполагает, что либо дочерний класс сохра-
няет поля и методы родительского класса, либо при необходимости изменяет 
их или дополняет новыми. При помощи наследования можно строить так назы-
ваемое дерево классов (или иерархию классов), последовательно уточняя описа-
ние класса и переходя от общих абстрактных понятий к частным. 

3. Полиморфизм – особый вид замены методов при наследовании, при ко-
тором программный код, работавший с методами родительского класса, приго-
ден для работы с измененными методами дочернего класса. 

Рассмотрим, как реализуется ООП в Object Pascal. Опишем некий класс 
для работы с домашними животными: 

type TPet = class 

       fName: string; // f – традиционный префикс для имени поля 

       fAge: Integer; 

       procedure SetAge(aAge: Integer); 

       function GetName: string; 

     end; 

Дадим необходимые комментарии. Описание класса размещается в сек-
ции описания типов. Описать класс можно в глобальной секции типов про-
граммы, в секции типов в интерфейсной части модуля или в разделе реализа-
ции модуля, но не в подпрограмме. Для описания класса используется ключе-
вое слово class. Вначале описываются поля класса. Полями класса TPet яв-
ляются fName и fAge. После описания всех полей следует указание методов 
класса. Методы TPet – это процедура SetAge и функция GetName. Класс со-
держит только заголовки методов, реализация методов описывается отдельно. 
Если класс описан в модуле, реализация методов должна находиться в том же 
модуле в секции implementation. Для указания того, что процедура или 
функция является реализацией метода некоего класса, используют синтаксис 
<имя класса>.<имя метода>. При реализации методов класса полностью 
доступны поля и другие методы этого класса без указания каких–либо допол-
нительных спецификаторов: 

procedure TPet.SetAge (aAge: Integer); 

begin 

  if aAge > 0 then fAge := aAge else fAge := 0 

end; 

function TPet.GetName: string; 

begin 

  Result := 'My name is ' + fName 

end; 

После того, как класс описан, можно объявить переменную класса, назы-
ваемую экземпляром класса или объектом: 

var Pet: TPet; 

Перед непосредственной работой с объектом он должен быть особым об-
разом инициализирован. Следующий код выполняет это: 

Pet := TPet.Create; 

После инициализации для работы с полями и методами объекта использу-
ется синтаксис вида <имя объекта>.<имя поля или метода>. Это напоминает 
работу с записью: 

Pet.fName := 'Tuzik'; 

Pet.SetAge(5); 

writeln(Pet.GetName+' I am ', Pet.fAge); 

Хотя прямая работа с полями объекта возможна, в ООП принято для дос-
тупа к полям использовать методы, подобные TPet.SetAge и TPet.GetName. 
Считается, что поля представляют внутреннюю закрытую часть класса, кото-
рую создатель класса вправе изменять. Заголовки методов доступа к полям 
(интерфейс класса) при этом должны оставаться неизменными. Object Pascal 

позволяет принудительно ограничить доступ к полям класса при помощи спе-
циальных директив. 

В методе TPet.SetAge содержится обращение к полю fAge. Каждый объ-
ект определенного класса содержит свой независимый набор полей. Метод 
SetAge можно вызвать для нескольких объектов. Возникает вопрос: как метод 
определяет, с полем какого объекта он работает? 

var P1, P2: TPet; 

. . . 

P1.SetAge(10); // SetAge работает с P1.fAge 

P2.SetAge(4); // SetAge работает с P2.fAge 

Для выявления конкретного объекта, с которым происходит работа, любо-
му методу передается скрытый параметр Self, который указывает на объект, 
вызывающий метод. Тип параметра Self совпадает с типом класса. На уровне 
компилятора описание метода SetAge и работу с ним можно представить сле-
дующим образом: 

procedure TPet.SetAge (aAge: Integer; Self: TPet); 

begin 

  if aAge > 0 then Self.fAge := aAge else Self.fAge := 0 

end; 

. . . 

P1.SetAge(10, P1); 

P2.SetAge(4, P2); 

Итак, Self указывает на текущий объект, с которым происходит работа. 
Практически всегда в явном использовании Self нет необходимости. Одно из 
исключений – использование одинаковых идентификаторов для полей класса и 
параметров метода. Предположим, что класс TPet содержит поле aAge, а не 
fAge. Тогда корректная реализация метода TPet.SetAge должна выглядеть так: 

procedure TPet.SetAge (aAge: Integer); 

begin  //aAge – параметр метода, Self.aAge – поле объекта 

  if aAge > 0 then Self.aAge := aAge else Self.aAge := 0 

end; 

4. Наследование классов. Перекрытие методов 
 
Предположим, возникла необходимость в описании класса, представляю-

щего домашних животных, способных издавать звуки. Подобный класс будет 
похож на класс TPet, но должен содержать некий дополнительный метод, от-
ветственный за воспроизведение звука. При помощи наследования можно опи-
сать новый класс на основе уже существующего: 

type TSpeakingPet = class(TPet) 

       procedure Speak; 

     end; 

. . . 

procedure TSpeakingPet.Speak; 

begin 

  Beep; 

end; 

Запись TSpeakingPet = class(TPet) означает, что класс TSpeakingPet 
является наследником класса TPet. Как наследник, TSpeakingPet содержит все 
поля и методы TPet и, кроме этого, добавляет собственный метод Speak. 

В Object Pascal все классы имеют одного общего предка. Таким предком 
является класс TObject. Объявления TPet = class и TPet = class(TObject) 
являются эквивалентными. Мы использовали один из методов класса TObject 
для инициализации объекта Pet (метод Create в классе TPet не описан). 

Объекты классов-потомков совместимы по присваиванию с объектами 
классов-предков. При этом действует следующее правило: объекту родитель-
ского класса можно присвоить объект дочернего класса, но не наоборот: 

var MyPet: TPet; SPet: TSpeakingPet; 

. . . 

MyPet := SPet; // допустимо 

SPet := MyPet; // ошибка компиляции 

Обосновывается вышеуказанное правило следующим образом: так как до-
черний класс может добавлять к родительскому новые поля, то при присваива-
нии объекту дочернего класса объекта родительского класса некоторые из по-
лей, возможно, не будут инициализированы. Это является недопустимым. 

Новый дочерний класс может не только дополнять родительский полями и 
методами, но и замещать методы родительского класса своими. Объявление в 
дочернем классе метода с тем же именем, что и в родительском классе называ-
ется перекрытием. При этом сигнатуры методов, т.е. количество и тип пара-
метров, могут отличаться. 

Для иллюстрации перекрытия объявим два новых класса, порожденных от 
TSpeakingPet: 

type TDog = class(TSpeakingPet) 

       procedure Speak; // перекрытие метода TSpeakingPet.Speak 

     end; 

     TCat = class(TSpeakingPet) 

       procedure Speak; // перекрытие метода TSpeakingPet.Speak 

     end; 

. . . 

procedure TDog.Speak; 

begin 

  writeln(‘Wuf’); 

end; 

procedure TCat.Speak; 

begin 

  writeln(‘Meau’); 

end; 

Для вызова перекрытых методов ближайшего класса-предка применяется 
конструкция inherited <имя метода класса–предка>. Если имя и параметры 
вызываемого у предка метода совпадают с именем и параметрами вызывающе-
го метода, достаточно записать inherited: 

procedure TCat.Speak; 

begin 

  inherited; // вызов TSpeakingPet.Speak – звуковой сигнал 

  writeln(‘Meau’); 

end; 

5. Конструкторы и деструкторы 
 
Object Pascal использует так называемую ссылочную объектную модель. 

Это означает, что все объекты размещаются в памяти динамически, объектные 
переменные фактически являются указателями на данные объекта в динамиче-
ской памяти и имеют одинаковый размер (4 байта). Однако для доступа к объ-
ектам не используется разыменователь «^» (мы записываем Pet.fName, а не 
Pet^.fName, хотя подразумевается именно второе). Для размещения объектов в 
динамической памяти служит особый вид методов, называемых конструкто-
рами. Для уничтожения объектов также предназначены особые методы – дест-
рукторы. 

Чтобы объявить метод как конструктор, используется ключевое слово 
constructor. Оно записывается вместо слова procedure в объявлении класса и 
при реализации метода (конструктор не может быть функцией). Компилятор 
автоматически добавляет к телу конструктора некий код, выделяющий для объ-
екта участок в динамической памяти. Так как конструктор необходимо выпол-
нить перед первым использованием объекта, то в тело конструктора обычно 
помещают операторы инициализации объекта, к примеру задание начальных 
значений для полей. В иерархии классов работа конструктора класса-потомка, 
как правило, начинается с вызова конструктора класса-предка для корректной 
инициализации полей предка. Отметим, что Object Pascal допускает существо-
вание в классе нескольких конструкторов. Традиционное имя для конструкто-
ра – Create. Класс TObject содержит конструктор Create, который обнуляет 
поля создаваемого объекта. 

Добавим конструктор в класс TPet: 
type TPet = class 

       . . . 

       constructor Create; 

     end; 

. . .  

constructor TPet.Create; 

begin 

  fAge := 1; 

  fName := ‘Pet’; 

end; 

Теперь оператор Pet := TPet.Create означает вызов не унаследованного 
конструктора TObject.Create, а собственного конструктора. Обратите внима-
ние на синтаксис вызова конструктора: <объектная переменная> := <имя 
класса>.<имя конструктора>. Как метод, конструктор можно вызывать и в 
виде <имя объекта>.<имя конструктора>  (Pet.Create). В первом случае 
конструктор работает как функция, которая выделяет память, размещает в ней 
объект соответствующего класса, выполняет тело конструктора и возвращает 

ссылку на выделенную область. Вызов, подобный Pet.Create, означает просто 
выполнение тела конструктора (реинициализацию полей). Его можно приме-
нять только для тех объектов, которые уже размещены в памяти. 

Для иллюстрации работы унаследованных конструкторов добавим конст-
руктор в класс TSpeakingPet: 

type TSpeakingPet = class(TPet) 

       constructor Create; // перекрытие конструктора 

     end; 

. . . 

constructor TSpeakingPet.Create; 

begin 

  inherited; // TPet.Create: fAge = 1, fName = ’Pet’ 

  fName := ‘Speaking Pet’; 

end; 

Для объявления деструкторов используется ключевое слово destructor. 
Задача деструктора – освободить память, которую занимал объект. Тело дест-
руктора – подходящее место для финальных действий с объектом (освобожде-
ние динамической памяти, закрытие файлов и тому подобное). Традиционное 
имя для деструктора – Desrtoy. Даже если ваш класс не содержит собственного 
деструктора, вызывайте унаследованный деструктор TObject.Destroy при 
окончании работы с объектом, чтобы не допускать утечек памяти: 

Pet := TPet.Create; // создание объекта 

Pet.fName := 'Tuzik'; // работа с объектом 

Pet.Destroy; // уничтожение объекта. Вызов TObject.Destroy 

Деструктор можно вызвать только у объекта, а не у класса. Попытка вы-
звать деструктор у неинициализированного объекта может привести к исклю-
чительной ситуации. Рекомендуется вместо прямого обращения к деструктору 
использовать метод TObject.Free, который вызывает деструктор Destroy, ес-
ли объект инициализирован (указатель на объект отличен от nil). Следует 
иметь ввиду, что метод Free вызывает не просто деструктор, а именно 
Destroy. Для корректной работы метода Free в унаследованных классах дест-
руктор должен называться Destroy (и быть виртуальным). 

6. Реализация полиморфизма 
 
Рассмотрим пример. Предположим, что имеется иерархия классов 

TPetTSpeakingPet(TDog, TCat). В программе необходимо создать массив 
из объектов класса TSpeakingPet или его наследников и для всех элементов 
массива вызвать метод Speak: 

var Zoo: array [1..3] of TSpeakingPet; 

. . .  

  Zoo[1] := TSpeakingPet.Create; 

  Zoo[2] := TDog.Create; // корректно по правилам присваивания 

  Zoo[3] := TCat.Create; // для объектов 

  for i := 1 to 3 do 

    Zoo[i].Speak; 

Ожидается, что результатом работы последнего цикла будут звуковой сиг-
нал и вывод строк ‘Wuf‘ и ‘Meau’. Однако в результате будет получена лишь 
последовательность из трех звуковых сигналов. 

Еще один пример. Опишем процедуру для обработки объектов класса 
TСryingPet: 

procedure Proc(A: TSpeakingPet); 

begin 

. . . 

  A.Speak; 

end; 

Хотя в Proc возможно передать не только объекты класса TCyingPet, но и 
любого дочернего класса, вызываться в теле процедуры будет метод 
TSpeakingPet.Speak. 

Подобные расхождения возникают из-за того, что нами до сих пор исполь-
зовался единственный вид методов – статический. Адрес статического метода 
известен на этапе компиляции. Он определяется классом объекта и не зависит 
от того, с каким классом будет связан объект на этапе выполнения. Нам необ-
ходимо, чтобы адрес метода вычислялся непосредственно в период выполнения 
по тому типу, который в этот момент имеет объект. Для этого необходимо при-
менить виртуальные методы. 

Работа с виртуальными методами происходит по следующей схеме. Каж-
дый объект наряду со значениями своих полей хранит указатель на специаль-
ную таблицу виртуальных методов (virual method table – VMT). Таблица вир-
туальных методов индивидуальна и единственна для каждого класса. В ней 
хранятся адреса всех виртуальных методов класса (как собственных, так и 
унаследованных). Связь между объектом и VMT класса осуществляется во вре-
мя начальной инициализации объекта – первого вызова конструктора. Вирту-
альные методы идентифицируются по константе-смещению в VMT. Во время 

выполнения программы из экземпляра объекта извлекается указатель на VMT 
и, используя константу-смещение, вычисляется адрес необходимого метода. 

Для объявления виртуального метода используется директива virtual. 
Методы, объявленные с директивой dynamic (динамические методы), функ-
ционально эквивалентны виртуальным. Иная внутренняя организация хранения 
адресов динамических методов позволяет сократить накладные расходы памя-
ти, но несколько замедляет работу с ними. 

Исходный метод, объявленный как виртуальный (динамический) в классе–
предке, перекрывается в классе–потомке методом с тем же именем и парамет-
рами, что и исходный, и помечается директивой override. Если хотя бы одно 
из этих требований не соблюдено, связь между виртуальными методами в ие-
рархии классов теряется. 

Отредактируем классы TSpeakingPet, TDog и TCat, сделав метод Speak 
виртуальным: 

type TSpeakingPet = class(TPet) 

       procedure Speak; virtual; 

     end; 

     TDog = class(TSpeakingPet) 

       procedure Speak; override; 

     end; 

     TCat = class(TSpeakingPet) 

       procedure Speak; override; 

     end; 

Если внести подобные изменения в описания классов, то вышеприведен-
ные примеры заработают корректно. При реализации методов директивы 
virtual, dynamic, override не указываются. 

Типичным примером полиморфизма в Delphi является объявление метода 
Repaint в классе TControl, предке всех визуальных компонентов. Благодаря 
тому, что этот метод объявлен как виртуальный, форма может прорисовать лю-
бой размещенный на ней визуальный компонент. 

Рассмотрим еще один пример, который требует применения виртуальных 
методов. Пусть имеется некий абстрактный класс TFigure для представления 
геометрических фигур. Его наследники TCircle и TSquare представляют собой 
соответственно круг и квадрат. Для рисования себя все три класса используют 
метод Draw, для стирания – метод Hide. Эти методы различны у всех классов. 
Кроме этого, TFigure содержит метод MoveTo для перемещения фигуры. Он 
состоит из вызова Hide, изменения координат фигуры и вызова Draw. Логика 
работы MoveTo полностью сохраняется для классов TCircle и TSquare. Однако 
если сделать методы Draw и Hide статическими, MoveTo придется дословно пе-
реписать в классах–потомках, так как TFigure.MoveTo содержит вызовы 
TFigure.Draw и TFigure.Hide. 

В этом примере для методов TFigure.Draw и TFigure.Hide реализация не 
важна. Главное то, что они делают, а не как. Можно оформить реализацию 
этих методов в виде пустых процедур. Лучшее решение состоит в указании ди-
рективы abstract, объявляющей метод абстрактным, не нуждающимся в реа-
лизации: 

type TFigure = class 

       . . . 

       procedure Draw; virtual; abstract; 

     end; 

Директива abstract применяется только для виртуальных и динамических 
методов в корневых классах объектных иерархий. Попытка создать экземпляр 
класса, содержащего абстрактные методы, вызовет предупреждение, а обраще-
ние к абстрактному методу сгенерирует исключительную ситуацию. 

Одним из ключевых понятий системы Windows является сообщение. Со-
общение – это сигнал приложению от операционной системы о том, что про-
изошло некоторое событие (к примеру, нажатие пользователем клавиши, пере-
мещение мыши, запуск приложения). Каждый класс сообщений идентифициру-
ется уникальной числовой константой. Object Pascal позволяет создавать мето-
ды для обработки сообщений. Схема внутренней реализации таких методов на-
поминает динамические методы. Метод обработки сообщений это всегда про-
цедура с единственным var-параметром. За заголовком метода следуют дирек-
тива message и номер сообщения Windows, которое этот метод будет обраба-
тывать. Этот номер обычно представляется мнемонической константой: 

type TCls = class 

       . . . 

       procedure WmUser(var M: TMessage); message wm_User; 

     end; 

Модуль Messages содержит константы для сообщений, а также большое 
количество типов, идентифицирующих конкретные сообщения. 

7. Свойства 
 
Развитие концепций ООП привело к появлению понятия свойства 

(property). Работа в программе со свойствами объекта происходит так же, как с 
полями объекта. Разница между полем и свойством заключается в следующем: 
обращение к свойству компилятор транслирует в вызов метода или обращение 
к полю, следовательно, при работе со свойствами могут выполняться некото-
рые действия (к примеру, при обращении к свойству «цвет» некоего объект, он 
перерисовывается указанным цветом); в отличие от полей свойства не занима-
ют места в памяти. 

Добавим свойства в класс TPet: 
type TPet = class 

       . . . 

       property Age: Integer read fAge write SetAge; 

       property Name: string read GetName write fName; 

     end; 

Для объявления свойства используется ключевое слово property. Далее 
следует имя свойства и указывается его тип. После директивы read указывает-
ся имя поля или метода для чтения свойства. Аналогично, после директивы 
write указывается имя поля или метода для записи свойства. Считается, что 
свойство записывается, когда ему присваивается некое значение, в противном 
случае свойство читается. Тип полей, используемых после read и write, дол-
жен совпадать с типом свойства. Метод, используемый для чтения простого 
свойства, должен быть функцией без параметров, тип возвращаемого значения 
которой совпадает с типом свойства. Метод для записи – процедура с одним 
параметром, имеющим тип свойства. Динамические методы и методы для об-
работки сообщений не могут использоваться для чтения и записи свойств. При-
нято имена методов чтения свойств начинать с префикса Get, имена методов 
записи свойств – с префикса Set. Объявление свойства может следовать только 
после объявления полей и методов, которые используются для чтения и записи. 
Работу со свойствами демонстрирует следующий фрагмент программы: 

var Pet: TPet; 

. . . 

Pet.Name := ‘Fish’; // оттранслируется в Pet.fName := ‘Fish’ 

Pet.Age := 1;       // Pet.SetAge(1); 

writeln(Pet.Age, Pet.Name); // writeln(Pet.fAge, Pet.GetName) 

То, что свойства не являются полями, накладывает определенные ограни-
чения на их использование. Свойства нельзя передавать в качестве var-
параметров в подпрограммы, к ним нельзя применить операцию взятия адреса. 

Если опустить директиву read, можно получить свойство только для запи-
си. Если опустить write, получим свойство только для чтения. Однако какая-то 
из директив должна присутствовать. 

Использование свойств является «хорошим тоном» ООП. Как было сказа-
но выше, следует избегать прямой работы с полями класса. В пользу использо-
вания свойств говорит тот факт, что в объектной библиотеке VCL весь доступ к 
полям классов и компонентов организован через свойства. 

Если класс имеет несколько «почти одинаковых» свойств, то для их обра-
ботки можно использовать одинаковые методы: 

type TExample = class 

       . . . 

       property Prop1: Byte index 0 read Get write Set; 

       property Prop2: Byte index 1 read Get write Set; 

       property Prop3: Byte index 2 read Get write Set; 

     end; 

Свойства Prop1, Prop2, Prop3 называются индексированными. Об этом го-
ворит использование директивы index с последующей уникальной целой кон-
стантой. Для чтения всех индексированных свойств применяется единственный 
метод-функция, содержащий один целый параметр. Запись всех индексирован-
ных свойств осуществляется также одним методом-процедурой, первый пара-
метр которого – целое число. Применение полей для чтения и записи индекси-
рованных свойств невозможно: 

var Ex: TExample; 

. . . 

Ex.Prop1 := 100; // Ex.Set(0, 100) 

k := Ex.Prop3; // k := Ex.Get(2) 

Object Pascal позволяет объявлять так называемые свойства-массивы, 
представляющие индексированное множество свойств: 

type TEx = class 

       . . . 

       property Props[i: Integer]: Byte read Get write Set; 

     end ; 

Как и для индексированных свойств, для доступа к свойствам-массивам 
возможно использование только методов. Первый параметр этих методов дол-
жен совпадать по типу с типом индекса в квадратных скобках. Ниже приведен 
пример работы со свойством Props и указано, во что оттранслируются обраще-
ния к нему: 

for i := 1 to 5 do 

  Ex.Props[i] := 0; //транслируются в Ex.Set(i, 0); 

Тип индекса свойства-массива не ограничен диапазоном (к примеру, ин-
декс может быть строкового или вещественного типа). Допускается работа 
только с элементами свойства-массива, а не со всем свойством целиком. 

Би
бл
ио
те
ка

 БГ
УИ
Р



 

 

Свойства-массивы могут быть многомерными. В этом случае количество 
необходимых параметров у методов чтения и записи увеличивается на соответ-
ствующее число. 

Если при описании свойства-массива добавить в конце директиву default, 
то такое- свойство становится основным свойством класса, что упрощает рабо-
ту с ним: 

property Props[i: Integer]: Byte read Get write Set; default; 

. . . 

for i := 1 to 5 do 

  Ex[i] := 0; //вместо Ex.Props[i] := 0 

Только свойство-массив может быть основным свойством класса. У класса 
может быть только одно основное свойство. 

8. Видимость атрибутов класса 
 
Под атрибутом класса далее понимается поле, метод или свойство класса. 
Как было указано выше, идеология ООП подразумевает, что не все данные 

и методы класса равноправны при использовании: одна их часть представляет 
внутреннюю организацию класса и должна быть скрыта от пользователя, дру-
гая служит интерфейсом класса и является общедоступной. 

Чтобы ограничить доступ к атрибуту класса, предусмотрено использова-
ние специальных директив ограничения видимости (мы пока их не использова-
ли). Они применяются не отдельно к каждому атрибуту, а разделяют объявле-
ние класса на секции: 

1. Private. Все атрибуты из данной секции доступны для использования 
только в том модуле, который содержит объявление класса. В этой секции 
обычно размещаются поля и методы, описывающие внутренние особенности 
реализации класса. 

2. Protected. Атрибуты этой секции могут использоваться вне пределов 
модуля с объявлением класса, но только потомками класса. В данную секцию 
обычно помещают виртуальные методы чтения и записи свойств. 

3. Public. Атрибуты из секции public не имеет ограничений на исполь-
зование. В эту секцию помещают методы и свойства, составляющие интерфейс 
класса. 

4. Published. Атрибуты секции published (публикуемые), как и атрибу-
ты из секции public, не имеют ограничений на использование. Компилятор ге-
нерирует для них специальную информацию, позволяющую работать с ними в 
IDE и Инспекторе объектов. В этой секции объявлено большинство свойств 
компонент. Тип поля из секции published может быть только классом, причем 
поддерживающим RTTI. RTTI (Run-time type information, информация о типе 
периода времени выполнения) – дополнительная информация об атрибутах 
класса, хранимая в классе и доступная во время выполнения программы. Класс 
с поддержкой RTTI либо откомпилирован с директивой $M+, либо является по-
томком класса, откомпилированного с этой директивой. Публикуемые свойства 
не могут быть свойствами-массивами. Тип публикуемого свойства может быть 
порядковым, множеством в пределах Integer, классом с поддержкой RTTI, Va-
riant, вещественным типом, любым строковым типом. Для таких свойств обя-
зательно наличие директивы как read, так и write. 

Если класс не содержит директив ограничения видимости, считается, что 
все его атрибуты имеют видимость public. Для классов с поддержкой RTTI (к 
примеру, класс TPersistent) по умолчанию принимается директива 
published. Порядок директив в описании класса произволен, допускается даже 
повторение директив. В отличие от C++, внутри модуля, содержащего описание 
класса, все директивы ограничения видимости не действуют. 

Би
бл
ио
те
ка

 БГ
УИ
Р



 

 

Для публикуемых свойств возможно указание спецификаторов default и 
stored, которые управляют процессом сохранения значения публикуемых 
свойств в DFM файле. 

Директива default служит для указания для свойства значения по умол-
чанию, которое должен устанавливать конструктор класса. Именно начальное 
значение отображается в Инспекторе объектов, и только если значение свойст-
ва отлично от принятого по умолчанию, оно сохранится в DFM-файле: 

type TC = class 

       . . . 

       published 

       property P: Integer read fP write fP default 5; 

     end; 

constructor TC.Create; 

begin 

  P := 5; 

end; 

Директива stored используется для указания того, будет ли свойство за-
поминаться в DFM-файле. В объявлении свойства за директивой stored должны 
следовать либо имя логического поля данного класса, либо логическая констан-
та, либо имя метода-функции класса, не имеющего параметров и возвращающе-
го логическое значение. Свойство запоминается в DFM-файле, если значение, 
следующее за stored, равно true: 

type TC = class 

       . . . 

       function NeedStore: Boolean; 

       published 

       property P: Integer 

                   read fP write fP stored NeedStore default 5; 

     end; 

Для свойств существует возможность изменения ограничения видимости в 
классе-потомке с protected на public или published. Для этого достаточно 
указать ключевое слово property и имя свойства в новой секции, не указывая 
тип свойства и методы или поля для чтения и записи. При необходимости мож-
но указать новое значение свойства по умолчанию: 

type TArrow = class(TGraphicControl) 

       . . .  

       published 

       property Height default 20; 

     end; 

9. Работа с классами. Обработчики событий 
 
Рассмотрим специальный тип данных, называемый метакласс. Этот тип 

введен для манипулирования классами, а не отдельными объектами. 
Описание метакласса и объявление соответствующей переменной выгля-

дят следующим образом: 
type TPetClass = class of TPet; 

var PetRef: TPetClass; 

На объявление метакласса указывает конструкция class of <имя клас-
са>. Что можно присваивать переменной PetRef? Ей присваивается либо имя 
класса, объявленного после class of, либо его дочерних классов: 

PetRef := TPet; 

. . . 

PetRef := TDog; 

Значением переменной типа метакласс фактически является указатель на 
VMT соответствующего класса. Так, в приведенном примере PetRef вначале 
содержит указатель на VMT класса TPet, затем указатель на VMT класса TDog. 

В Object Pascal имеется предопределенный тип TClass: 
type TClass = class of TObject; 

Так как TObject является предком для всех объектов, то переменной типа 
TClass можно присваивать значение любого класса. 

Ценность метаклассов состоит в возможности создавать программы, кото-
рые манипулируют классами, не существующими в момент написания про-
граммы. Примером этого служит работа с формой. Когда мы конструируем 
форму, мы создаем новый класс, порожденный от TForm. Класс TForm содержит 
виртуальный конструктор. Метод TApplication.CreateForm ответственен за 
создание и визуализацию формы. Его заголовок имеет вид TApplica-

tion.CreateForm(T: class of TForm; var F: TForm). В теле метода созда-
ется объект F класса T. Благодаря ссылке на класс и виртуальному конструкто-
ру метод CreateForm может создать объект не только класса TForm, но и любо-
го дочернего класса. 

 
9.1. Классовые методы 

 
В предыдущих примерах для вызова метода класса использовался экземп-

ляр класса. Object Pascal позволяет описать в классе такие методы, для вызова 
которых нет необходимости создавать объект. Эти методы называются классо-
выми (class method). Для их объявления достаточно указать слово class непо-
средственно перед procedure или function: 

type TSomeClass = class 

       class procedure Hello; 

     end; 

class procedure TSomeClass.Hello; 

begin 

  writeln(‘Hello. This is class method’) 

end; 

Так как в момент вызова классового метода экземпляр класса может и не 
существовать, то в теле такого метода запрещено обращение к полям объекта и 
обычным методам. Однако классовый метод может вызывать другие классовые 
методы и конструктор класса. 

Для вызова классового метода возможно применение как синтаксиса <имя 
класса>.<имя классового метода>, так и <имя объекта>.<имя классового 
метода>. Для вызова классового метода можно применять метакласс: 

type TSomeClassRef = class of TSomeClass; 

var C: TSomeClass; 

    R: TSomeClassRef; 

. . . 

  TSomeClass.Hello; // все вызовы корректны 

  C := TSomeClass.Create; //дают одинаковый результат 

  C.Hello; 

  R := TSomeClass; 

  R.Hello; 

Указатель Self для классового метода содержит адрес VMT класса. Раз-
решено объявление виртуальных классовых методов. Для классовых методов 
действуют обычные правила видимости атрибутов класса. 

 
9.2. Указатели на методы. Обработчики событий 

 
В Object Pascal, как и в Turbo Pascal, есть процедурные типы: 

type Proc = procedure (k: Integer); 

var P: Proc; 

Однако, хотя метод TPet.SetAge имеет такую же сигнатуру, как и при 
объявленнии типа Proc, присваивание P := Pet.SetAge, где Pet – экземпляр 
TPet, является недопустимым. Причина этого в том, что методу всегда переда-
ется неявный параметр Self. Процедурные типы для методов называются ука-
зателями на метод (method pointer) и объявляются следующим образом: 

type Proc = procedure (k: Integer) of object; 

var P: Proc; 

Конструкция of object указывает на объявление указателя на метод. Те-
перь возможно присваивание вида P := Pet.SetAge. Переменная типа указа-

тель на метод занимает 8 байт и содержит адрес метода и ссылку на объект 
(значение Self для конкретного объекта). 

Указатели на метод используются для создания и манипулирования собы-
тиями. Любое событие объекта представляет собой свойство типа указатель на 
метод. IDE Delphi автоматически помещает все такие свойства класса, объяв-
ленные в секции published, на закладку Events Инспектора объектов. Тип 
простейших событий с единственным параметром Sender описан следующим 
образом: 

type TNotifiEvent = procedure(Sender: TObject) of object; 

В качестве примера рассмотрим класс-компонент TЕdit. Он обладает со-
бытием OnClick. Для реализации работы с ним класс содержит поле FОnClick 
и свойство OnClick: 

type TEdit = class(TCustomEdit) 

       . . . 

       FOnClick: TNotifiEvent; 

       property OnClick: TNotifiEvent 

                         read FOnClick write FOnClick; 

     end; 

Если в процессе разработки мы размещаем на форме компонент Edit1: 
TEdit и пишем для него обработчик события OnClick, то фактически мы пи-
шем новый метод класса формы TForm1.Edit1Click. При создании формы в 
начале выполнения приложения метод формы связывается со свойством ком-
понента следующим образом (это происходит без нашего участия): 

Edit1.OnClick := Form1.Edit1Click; // Form1 – объект TForm1 

При щелчке на компоненте Edit1 системой вызывается метод 
TEdit.Click. В нем производятся проверка наличия обработчика события 
OnClick и вызов его, если он существует: 

if Assigned(OnClick) then OnClick(Self); 

Функция Assigned возвращает значение «истина», если ее аргумент не ра-
вен nil. 

Подобный подход, когда все методы обработки событий компонент сосре-
доточены в одном классе (TForm1), называется делегированием. Делегирование 
позволяет избежать порождения от существующих классов-компонентов мно-
гочисленных потомков, содержащих дополнительные методы для обработки 
событий. 

10. Приведение и контроль типов объектов 
 
Любой класс содержит некую дополнительную информацию, размещае-

мую в памяти непосредственно перед VMT. Эти данные позволяют во время 
выполнения программы контролировать (type checking) и приводить (type 
casting) объектные типы. 

Для контроля типов используется оператор is. Выражение <объект> is 
<класс> возвращает «истину», если <объект> принадлежит классу <класс> 
или потомкам этого класса (if Obj1 is TPet then . . .). 

Для приведения типов используется оператор as в следующей форме: 
(Obj1 as TPet).SetAge(10); 

Допустима традиционная конструкция TPet(Obj1).SetAge(10), однако 
вариант с as является более безопасным. В случае неудачи он генерирует обра-
батываемую исключительную ситуацию, а жесткое приведение типов 
TPet(Obj1) может привести к краху приложения. 

Приведение и контроль типов можно использовать для написания универ-
сальных обработчиков событий. Предположим, на форме имеются компоненты 
TEdit и TButton. Необходимо при щелчке на компоненте TEdit установить 
свойство Text=’Hi’, при щелчке на TButton – свойство Caption = ‘Hello’: 

procedure TForm1.Edit1Click(Sender: TObject); 

begin 

  if Sender is TEdit then 

    TEdit (Sender).Text := 'Hi' 

  else if Sender is TButton  then 

         TButton(Sender).Caption := 'Hello' 

end; 

Приведенный обработчик события TEdit.Click решает эту задачу. Этот 
обработчик нужно назначить событию TButton.Click, выбрав его в выпадаю-
щем списке на закладке Events в Инспекторе объектов. В примере использо-
вался тот факт, что любой обработчик события имеет параметр Sender, указы-
вающий на объект, который инициировал событие. 

11. Исключительные ситуации 
 
Термин исключительная ситуация (exception) обозначает любую ошибку 

или ошибочное условие, возникающие в процессе выполнения программы. 
Примерами исключительных ситуаций являются деление на 0, запись в файл, 
открытый только для чтения, ошибки при преобразовании типов. Традицион-
ный подход для контроля и обработки подобных ошибок заключается в исполь-
зовании условных операторов. Однако они загромождают программу, делают 
ее менее ясной. 

Object Pascal предлагает новый подход: при исключительной ситуации по-
рождается объект особого класса, при помощи этого объекта информация об 
исключительной ситуации передаётся специальному обработчику, написанному 
программистом. 

Каждому типу исключительной ситуации соответствует некий класс. Эти 
классы образуют иерархию, корнем которой является класс Exception. Приве-
дем некоторые классы и соответствующие им исключительные ситуации: 

EAbort – «безмолвная» исключительная ситуация, без сообщений; генери-
руется вызовом процедуры Abort; 

EDivByZero – деление целого числа на 0; 
EInOutError – ошибка при доступе к файлу; 
EOutOfMemory – нехватка памяти; 
EMathError – общая ошибка при работе с вещественными числами; 
EZeroDivide – деление вещественного числа на 0; 
EConvertError – ошибка при преобразовании типов. 
Полный список классов содержит справочная система Delphi. Основные 

классы исключительных ситуаций описаны в модуле SysUtils. 
Для генерации объекта исключительной ситуации применяется предложе-

ние raise. Вот как выглядит создание объекта в случае нехватки памяти: 
raise EOutOfMemory.Create('Мало памяти'); 

Обратите внимание: вызывается конструктор класса, соответствующего 
исключительной ситуации. Для гибкой обработки исключительных ситуаций 
класс Exception предлагает несколько конструкторов. Мы использовали про-
стейший, параметр которого устанавливает свойство Message класса 
Exception. 

Обработчики исключительных ситуаций реализуются при помощи защи-
щенного блока try-except. Синтаксис блока следующий: 

try 

  //операторы, которые могут вызвать исключительную ситуацию 

except 

  //операторы обработки исключительных ситуаций 

end; 

При нормальном ходе программы выполняются операторы, размещенные 
между try и except, а затем – размещенные после end. Если какой–либо из 
операторов между try и except вызвал исключительную ситуацию, управле-
ние сразу передается в часть except-end. Предполагается, что в ней сосредото-
чена обработка исключительной ситуации. После обработки исключительной 
ситуации выполняются операторы, размещенные за end. Конструкции try-
except могут быть вложенными. 

Для распознавания исключительной ситуации в части except-end служит 
последовательность блоков обработки вида on <класс исключительной си-
туаций> do <оператор>. К примеру: 

try 

  //работа с вещественными числами 

except 

  on EZeroDivide do . . . ; //обрабатываем ошибку деления на 0 

  on EMathError do . . . ; //обработка других матем. ошибок 

end; 

Если для класса исключительной ситуации существует блок обработки, 
выполняется часть блока после do и управление передается за блок try-
except. Порядок блоков обработки имеет значение: вначале проводится распо-
знавание частных исключительных ситуаций, затем – более общих (как и в при-
веденном примере). 

Если необходимо, чтобы нераспознанные исключительные ситуации были 
как-то обработаны, можно поместить операторы их обработки в секцию 
except-end после слова else: 

except 

  on EZeroDivide do . . . ; //обрабатываем ошибку деления на 0 

  on EMathError do . . . ; //обработка других матем. ошибок 

  else //обработка нематематических ошибок 

end; 

Приведем пример функции, генерирующей исключительную ситуацию: 
function StrToPersent(const S: string): Integer; 

begin 

  Result := StrToInt(S); 

  if (Result < 0) or (Result > 100) then 

    raise EConvertError.Create(S + 'is not a valid value') 

end; 

Функция StrToPersent может генерировать исключительную ситуацию клас-
са EConvertError. В следующей функции она распознается и обрабатывается: 

function IncPercent(const S: string): string; 

begin 

  try 

    Result := IntToStr(StrToPercent(S) + 1); 

  except 

    on EConvertError do Result := '0'; 

  end 

end; 

Если мы хотим возложить часть обработки исключительной ситуации на 
некий внешний блок, то исключительную ситуацию можно сгенерировать по-
вторно, используя raise: 

on EZeroDivide do 

begin 

 . . . //частично обработали мы 

  raise //пусть обрабатывает внешний обработчик 

end; 

Если необходимо получить доступ к объекту, описывающему исключи-
тельную ситуацию, то используется следующая форма блока обработки: on 
<идентификатор объекта>: <класс исключительной ситуации> do <опе-
ратор>; 

on E: EConvertError do  

begin 

  Result := '0'; 

  ShowMessage(E.Message) 

end; 

Освобождение созданного объекта исключительной ситуации происходит 
автоматически. Переменную, используемую в блоке on (в примере это Е), нигде 
описывать не надо. 

Для корректного освобождения ресурсов приложения часто используется 
блок try-finally. Синтаксис блока следующий: 

try 

  // операторы, которые могут вызвать исключительную ситуацию 

finally 

  // эти операторы выполняются всегда 

end; 

При возникновении исключительной ситуации в части try-finally 
управление передается на часть finally-end. При нормальной работе выпол-
няются все операторы между try и end. 

Если возникшая исключительная ситуация программистом не обработана, 
вызывается обработчик события OnExcept объекта Application. Если этот об-
работчик не установлен, выводится стандартное окно с текстовым сообщением 
об ошибке. 

12. Иерархия классов VCL 
 
Объектная библиотека VCL насчитывает около 200 классов, составляющих 

сложное дерево иерархии. Основными классами, «стволом дерева», являются 
следующие: 

Tobject  Tpersistent  Tcomponent  Tcontrol  (Tgraphic-

Control, TWinControl) 

Класс TObject является общим предком всех классов. Он содержит конст-
руктор Create, виртуальный деструктор Destroy, метод Free для освобожде-
ния объекта, методы для работы с сообщениями системы Dispatch и 
DefaultHandler и большое число классовых методов для работы с информаци-
ей о классе. Эти методы позволяют узнать предок класса (ClassParent), имя 
класса (ClassName), адрес метода в секции published по имени метода 
(MethodAddress), размер объекта данного класса (InstanceSize) и многое дру-
гое. Виртуальные методы TObject хранятся по отрицательным смещениям таб-
лицы VMT, так что «первым» методом в VMT будет первый виртуальный ме-
тод производного класса. Классы Exeption, TList, TPrinter порождены непо-
средственно от класса TObject. 

Особенностью класса TPersistent является то, что он и его потомки со-
держат методы для сохранения и чтения данных своих полей. TPersistent со-
держит также методы для копирования данных из полей одного объекта в поля 
другого. Так как объекты являются всего лишь указателями на данные в памя-
ти, то код 

var A, B: TSomeClass; 

. . .  

A := B; 

приведет к тому, что и A и B будут ссылаться на один участок памяти, а данные, 
связанные с объектом A, будут потеряны. Чтобы этого не произошло, следует 
использовать метод Assign, который позволяет вызвавшему объекту присваи-
вать данные, связанные с другим объектом, и метод AssignTo, копирующий 
данные вызвавшего объекта в другой объект. На уровне TPersistent данные 
методы объявлены как абстрактные, классы–потомки должны переопределить 
их. Класс TPersistent скомпилирован с директивой $M+, а значит, у него и у 
всех его потомков атрибуты по умолчанию имеют видимость published. Клас-
сы TString, TCanvas, TGraphic, TPicture порождены непосредственно от 
TPersistent. 

Рассмотрим подробнее класс TStrings. Он относится к классам общего 
назначения. Задачей классов общего назначения является реализация типичных 
структур данных – списков, потоков, коллекций. Такие классы можно исполь-
зовать самостоятельно, но задумывались они как вспомогательные элементы 

крупных классов VCL. Большая часть классов общего назначения описана в 
модуле Classes. 

Класс TStrings служит для представления списка строк. Этот класс явля-
ется абстрактным, непосредственно оперировать его методами нельзя. Многие 
компоненты имеют свойства, тип которых представляет класс, порожденный от 
TStrings. Класс TStringList является наследником TStrings и допускает 
прямое использование. 

Перечислим основные свойства класса TStrings: 
Count: Integer – число элементов в списке; 
Strings[Index: Integer]: string – массив строк, основное свойство, 

элементы нумеруются с нуля; 
Objects[Index: Integer]: TObject – массив объектов, это свойство по-

зволяет ассоциировать с каждой хранимой строкой некий объект; 
Text: string – позволяет интерпретировать массив строк как одну боль-

шую строку с разделителями #13#10. 
Основные методы класса TStrings приведены ниже: 
Add – добавляет строку в массив и возвращает её позицию; 
AddObject(const S: string; AObject: TObject): Integer – добавля-

ет в массив строку S и ассоциирует с ней объект AObject; 
AddStrings(St: TStrings) – добавляет в массив список строк; 
Append – аналог Add, но является процедурой; 
Clear – удаляет все элементы списка; 
Delete(Index: Integer) – удаляет строку с индексом Index и ассоции-

рованный с ней объект. Delete и Clear не вызывают у удаляемых объектов де-
структор; 

Equals(String: TStrings): Boolean – сравнивает два списка; 
Exchange(Ind1. Ind2: Integer) – меняет два элемента списка местами; 
GetText: PChar – возвращает весь список в виде строки с завершающим 

нулем; 
IndexOf(const S: string ): Integer – позиция строки S в списке 

или –1; 
Insert(Index: Ineger; const S: string ) – вставка в список строки в 

заданную позицию; 
LoadFromFile(const Name: string), SaveToFile(const Name: string) – 

загрузка или сохранение списка в файле; 
Move(CurIndex, NewIndex: Integer) – перемещение элемента списка; 
SetText(Text: PChar) – загрузка списка из одной строки с завершающим 

нулем. 
Класс TStringList добавляет к TStrings несколько новых свойств, мето-

дов и событий. 
Свойства класса TStringList: 
Duplicates – определяет, разрешено ли использование дубликатов в спи-

ске. Возможные значения: DupIgnore – дубликаты игнорируются, DupAccept – 

дубликаты разрешаются, DupError – попытка добавить дубликат вызывает 
ошибку; 

Sorted: Boolean – при установке в true строки автоматически сортиру-
ются по алфавиту. 

Методы класса TStringList: 
Find(const S: string; var Index: Integer): Boolean – ищет строку 

S, если она найдена, возвращает true и индекс строки; 
Sort – сортирует строки списка по алфавиту. 
События класса TStringList: 
OnChange – генерируется после внесения изменений в список; 
OnChanging – генерируется перед внесением изменений в список. 
Класс TComponent происходит от TPersistent и является предком визу-

альных и невизуальных компонентов. Объекты класса TComponent называются 
компонентами. На уровне TComponent реализована поддержка возможности 
редактирования свойств компонентов в Инспекторе объектов. Компоненты 
также поддерживают возможность владения другими компонентами, при этом 
компонент-владелец становится ответственным за уничтожение принадлежа-
щих ему компонентов. Некоторые из свойств класса TComponent перечислены 
ниже: 

Owner – компонент, который владеет данным; 
ComponentState – текущее состояние компонента; 
ComponentCount – количество компонентов, принадлежащих данному; 
Components – массив принадлежащих компонентов; 
ComponentIndex – текущая позиция в этом массиве; 
Name – имя компонента, используется как имя поля в классе формы; 
Tag – число типа Integer. 
Класс TComponent определяет виртуальный конструктор Create с одним 

параметром – именем компонента-владельца создаваемого компонента. 
Методы TComponent позволяют найти компонент по имени в массиве 

Components (FindControl), получить список всех компонент, владельцем ко-
торых является данный, и так далее. Класс TScreen и большинство классов не-
визуальных компонентов порождены непосредственно от TComponent. 

Объект Screen класса TScreen инкапсулирует информацию об экране. 
Этот объект создается автоматически в начале работы приложения. Большин-
ство свойств класса TScreen доступно только для чтения. Некоторые из 
свойств перечислены ниже: 

ActiveForm – форма приложения, имеющая фокус ввода; 
ActiveControl – элемент управления, имеющий фокус; 
Cursor – вид курсора на экране; 
Cursors – массив доступных курсоров; 
FormCount – число форм, отображённых на экране; 
Forms – массив видимых форм программы; 
Fonts: TStrings – имена системных шрифтов для вывода текста на 

экран; 

Width, Height – ширина и высота экрана в пикселях. 
Наследником класса TComponent является класс TApplication. Задача 

этого класса – инкапсулирование поведения приложения в целом. В любом 
приложении существует автоматически создаваемый объект Application этого 
класса. 

Основные свойства класса TApplication приведены ниже: 
MainForm: TForm – главная форма приложения; 
ExeName – строка, содержит пути и имя приложения; 
Title – название приложения на панели задач; 
HelpFile – имя файла-справки для приложения; 
Icon – значок приложения на панели задач; 
Hint – текст всплывающей подсказки; 
HintColor – цвет окна всплывающей подсказки; 
Handle – дескриптор приложения. 
Некоторые события класса Tapplication: 
OnActivate – приложение стало активным; 
OnDeactivate – приложение теряет фокус при переключении приложений; 
OnException – возникла исключительная ситуация, которая не обработа-

лась; 
OnIdle – приложение простаивает. 
Приведем некоторые из методов класса Tapplication: 
Terminate – завершает работу приложения; 
ProcessMessages – обрабатывает системные сообщения Windows; исполь-

зуется в теле длинных циклов для устранения эффекта «замораживания»; 
ShowException(E: Exception) – выводит окно с описанием исключи-

тельной ситуации E; 
Minimize – свёртывает все окна приложения; 
Restore – развёртывает все окна приложения; 
MessageBox – показывает простейшее диалоговое окно с текстом; 
CreateForm – создает и отображает форму. 
От класса TComponent происходит класс TControl – общий предок визу-

альных компонентов. Класс TControl содержит свойства позиционирования 
компонента (Top, Left, Widht, Height, Align), свойства клиентской области 
(ClientRect, ClientWidth, ClientHeight), свойства внешнего вида (Visible, 
Enabled, Font, Color), строковые свойства (Caption, Text, Hint), свойства 
мыши (Cursor, DragCursor, DragMode). На уровне класса TControl появляют-
ся обработчики событий мыши (OnClick, OnDblClick, OnMouseDown, 
OnMouseMove, OnMouseUp) и событий перетаскивания Drag&Drop (OnDragOver, 
OnDragDrop, OnEndDrag). Каждый компонент класса TControl имеет родитель-
ский компонент класса TWinControl, указываемый в свойстве Parent и отве-
чающий за отображение компонента. Если свойство Parent не установлено, 
компонент не отображается. 

Класс TWinControl является предком для компонентов, инкапсулирующих 
стандартные элементы управления, которые могут получать фокус (поля ввода, 

кнопки, списки и другие). Компоненты класса TWinControl могут быть роди-
тельскими (Parent) для других визуальных компонентов. Основные свойства 
класса предназначены для управления внешним видом компонента и фокусом 
ввода: 

Handle – дескриптор, целое число для идентификации объекта системой; 
Ctrl3D – определяет, выглядит ли компонент объемным; 
HelpContext – раздел справочной системы для оперативной подсказки; 
Controls – список элементов управления, для которых компонент являет-

ся родительским; 
TabStop, TabOrder – управляют перемещениями по компонентам на фор-

ме при нажатии клавиши Tab. 
Класс TWinControl содержит методы управлением фокусом (Focused, 

CanFocus, SetFocus), методы позиционирования и выравнивания, методы соз-
дания и уничтожения окна (CreateParams, CreateWnd, DestroyWnd). Метод 
ControlAtPos возвращает дочерний компонент для указанной точки, метод 
FindNextControl позволяет найти компонент, получающий фокус после дан-
ного компонента. 

Компоненты класса TWinControl поддерживают обработку событий кла-
виатуры (OnKeyDown, OnKeyUp, OnKeyPress) и событий управления фокусом 
(OnEnter, OnExit). 

Класс TGraphicControl используется для визуальных компонентов, кото-
рые отображаются (рисуются) системой Delphi, а не Windows. Это позволяет 
экономно расходовать системные ресурсы. Потомки класса TGraphicControl 
(к примеру, TLabel) не могут получать фокус. Класс TGraphicControl содер-
жит свойство Canvas – «холст» для рисования компонента и виртуальный ме-
тод Paint, содержащий операторы для рисования компонента и переопреде-
ляемый потомками класса. 

В качестве примера работы с описанными классами рассмотрим, как соз-
дать компонент во время выполнения программы: 

procedure TForm1.Button1Click(Sender: TObject); 

var B: TButton; 

begin 

  B := TButton.Create(Self); //владелец новой кнопки – форма 

  B.Top := 20; //кнопка невидима, настроим ее внешний вид 

  B.Left := 40; 

  B.Parent := Self; //сейчас кнопка отобразилась на форме 

end; Би

13. Создание и использование DLL 
 
Библиотека динамической компоновки (dynamic link lybrary, DLL) – вы-

полняемый модуль Windows, код и ресурсы которого могут использоваться 
приложениями и другими DLL. Файлы библиотек динамической компоновки 
обычно имеют расширение DLL. Применение DLL оправдано, если приложение 
разрабатывается на нескольких языках программирования, программа содержит 
национальные ресурсы (к примеру, строки сообщений), существует несколько 
приложений, использующих сходный набор функций. 

 
13.1. Структура DLL 

 
Структура DLL схожа со структурой обычной программы. Она начинается 

с ключевого слова library, за которым следует имя библиотеки: 
library MyDLL; 

Все подпрограммы, которые будут вызываться из других приложений, 
объявляются в DLL с директивой stdcall. Вызов подпрограмм, объявленных 
без директивы stdcall, может быть затруднен в приложениях, написанных не 
в Delphi: 

procedure MyProc(A: Integer; B: Char);stdcall; 

Подпрограммы, экспортируемые из DLL, должны быть перечислены в 
специальном разделе exports исходного кода: 

exports MyProc, MyFunct; 

Таких разделов может быть несколько и в любом месте исходного кода 
DLL, главное, чтобы экспортируемые подпрограммы были описаны перед упо-
минанием в разделе exports. 

Каждая экспортируемая из DLL подпрограмма идентифицируется для рас-
познавания двумя ключами: числовым индексом и именем экспорта. По умол-
чанию имя экспорта совпадает с именем подпрограммы (большие и малые бук-
вы в имени экспорта различаются), а индекс – с порядковым номером подпро-
граммы в разделе exports. Однако такой порядок можно обойти, явно указав 
имя экспорта и (или) числовой индекс: 

exports 

  MyProc1, 

  MyProc2 name 'MyDinProc', 

  MyProc3 index 3, //целое число от 1 до 2,147,483,647 

  MyProc4 index 4 name 'MyPr@1'; //в имени допустимы спецсимволы 

Будьте внимательны при явном указании индекса – его уникальность ком-
пилятором не отслеживается. 

Рассмотрим пример DLL. Для получения некой «заготовки» DLL в IDE 
можно использовать следующую последовательность команд: File|New…|DLL. 
Далее наполняем эту «заготовку» содержимым: 

library MathLib; 

uses SysUtils; //модуль Classes нами не используется 

function Max(X,Y: Integer): Integer;stdcall; 

begin 

  if X < Y then Result := Y else Result := X 

end; 

function Min(X,Y: Integer): Integer;stdcall; 

begin 

  if X < Y then Result := X else Result := Y 

end; 

exports Min, Max; 

begin 

end. 

После компиляции этого кода (Ctrl+F9) в рабочем каталоге получим файл 
Mathlib.dll. 

Между begin..end в коде DLL можно поместить операторы инициализа-
ции. Они выполняются при загрузке библиотеки DLL в память. 

 
13.2. Импорт подпрограмм из DLL 

 
Различают следующие виды импорта подпрограмм DLL – статический и 

динамический. 
При статическом импорте DLL загружается в память в момент старта при-

ложения и находится там до окончания его работы. При этом в исходном коде 
приложения размещены явные ссылки на подпрограммы DLL. Следующие 
примеры демонстрируют три возможных вида этих ссылок: 

1. По имени подпрограммы в исходном коде DLL: 
procedure MyProc1;stdcall;external 'MYLIB.DLL'. 

2. По имени экспорта: 
procedure MyProc2;stdcall;external 'MYLIB.DLL' name 

'MyDinProc2'. 

3. По числовому индексу: 
procedure MyProc3;stdcall;external 'MYLIB.DLL' index 3. 

Рекомендуется использовать импорт по имени экспорта, он более нагля-
ден, хотя и работает немного медленнее, чем импорт по числовому индексу. 

При динамическом импорте подпрограмм DLL загружается в память и вы-
гружается самим приложением по мере надобности. Динамическая загрузка 
происходит следующим образом. Вначале вызывается функция LoadLibrary 
(LibFileName: PСhar): HМodule, где LibFileName – имя файла DLL. Если 
указано имя файла без маршрута, то DLL ищется в каталоге, из которого запу-
щено приложение, затем – в текущем каталоге, системном каталоге 
(Windows\System32), каталоге Windows, каталогах, указанных в системной пе-
ременной PATH. Функция возвращает системный дескриптор DLL (число) или 
код ошибки от 0 до 32: 

var LibHandle: THandle; //встроенный тип для дескрипторов 

begin 

. . . 

  LibHandle := LoadLibrary('MathLib.dll'); 

  if LibHandle < 32 then . . .//обработка ошибки 

. . . 

end; 

После загрузки DLL для получения адресов подпрограмм используется 
функция GetProcAddress(Module: HModule; ProcName: PChar): Pointer, 
где ProcName – имя подпрограммы. Если мы хотим произвести поиск подпро-
грамм по индексу, то два младшие байта указателя PChar должны содержать 
индекс, а два старшие – ноль. Если подпрограмма не найдена, при использова-
нии имени GetProcAddress возвращает nil, при использовании индекса значе-
ние GetProcAddress может быть отлично от nil. 

При динамическом импорте DLL в вызывающей программе необходимо 
объявить соответствующие процедурные типы. Приведем пример динамиче-
ского импорта: 

type TMin = functuion (X, Y: Integer): Integer;stdcall; 

. . .  

var Min: TMin 

. . . 

  @Min := GetProcAddress(LibHandle, 'Min'); 

  A := Min(B,C); 

После завершения работы с библиотекой она освобождается вызовом 
функции FreeLibrary(LibModule: HModule): BOOL, где LibModule – деск-
риптор DLL. Функции LoadLibrary, GetProcAddress, FreeLibrary объявлены 
в модуле Windows. 

Достоинствами динамического импорта DLL являются эффективное ис-
пользование ресурсов памяти и возможность контролировать и обрабатывать 
ситуации, когда необходимой подпрограммы в DLL не оказалось. 

13.3. Ресурсы в DLL 
 
DLL может содержать не только подпрограммы, но и ресурсы, в частности, 

описание одной или нескольких форм. Для помещения формы в DLL во время 
разработки DLL в IDE необходимо выполнить команду File|New form. Затем 
новая форма настраивается необходимым образом (как правило, форма пред-
ставляет диалоговое окно). Далее в исходный файл DLL добавляется процеду-
ра, которая показывает эту форму: 

procedure ShowTheForm(AOwner: Tcomponent);stdcall; 

begin 

  Form1 := TForm1.Create(AOwner); 

  Form1.ShowModal; 

  Form1.Free 

end; 

. . .  

exports ShowTheForm; 

Эта процедура используется в основной программе: 
procedure ShowTheForm(AOwner :TComponent);stdcall;external 

'MYDLL.DLL'; 

. . .  

ShowTheForm(Form1); 

 
13. 4. Некоторые замечания 

 
Если DLL экспортирует подпрограммы, использующие в качестве пара-

метров значения типа Variant, длинные строки и динамические массивы, эта 
DLL и использующее ее приложение должны подключать модуль ShareMem 
(причем первым в списке uses) и иметь доступ к библиотекам DELPHIMM.DLL 
или BORLANDMM.DLL (поставляются в комплекте с Delphi). 

Если в подпрограмме DLL возникла исключительная ситуация, то соответ-
ствующий объект передается в вызывающее приложение. Обработка исключи-
тельной ситуации возможна, если DLL использует модуль SysUtils. 

Если в DLL объявлены свои глобальные переменные и константы, то дос-
туп к ним из вызывающей программы возможен только через процедурный ин-
терфейс. 

Для главной программы, использующей DLL, обычно создается дополни-
тельный интерфейсный модуль. В секции interface модуля указываются тре-
буемые подпрограммы с директивой stdcall, а в секции implementation пе-
речисляются с директивой external экспортируемые из DLL процедуры. 

14. Разработка пользовательских компонентов 
 
Одной из ключевых особенностей Delphi, хорошо иллюстрирующей пре-

имущества объектно-ориентированного программирования, является возмож-
ность создания и использования новых пользовательских компонентов. При 
этом благодаря наследованию процесс создания компонента никогда не начи-
нается «с нуля», а благодаря полиморфизму компонент органически встраива-
ется в IDE и библиотеку VCL. 

Создание компонента – невизуальный процесс. Он включает следующие 
стадии: 

1. Написание программного кода компонента. 
2. Создание значка компонента для Палитры компонентов и файла справ-

ки компонента (необязательный этап). 
3. Инсталляция компонента. 
Процесс создания компонента в различных версиях Delphi может иметь 

небольшие отличия. Следующее описание опирается на версию Delphi 3. 
 

14.1. Компонент TNewLabel 
 

Проиллюстрируем этапы создания компонента на примере TNewLabel – 
надписи, которая имеет отдельное свойство для изменения цвета шрифта. 

Для получения модуля с «заготовкой» программного кода будущего ком-
понента служит команда IDE File|New|Component. Заполним в появившемся 
диалоге следующие поля: Ancestor Type – TLabel, Class Name – TNewLabel, 
Pallete Page – Samples, Unit File Name – ‘c:\work\newlabel’. 

Дадим необходимые комментарии. Ancestor Type – класс предка. Если 
новый компонент дополняет возможности существующего, указывается класс 
существующего компонента (как в нашем случае). Если компонент по содер-
жанию радикально новый, то указываем: TComponent – для невизуальных ком-
понентов; TCustomControl – для компонентов, которые должны получать фо-
кус; TGraphicControl – для рисуемых компонентов; TCustomXXX 

(TCustomLabel) – если компонент дополняет возможности существующего, но 
желательно уменьшить количество свойств и событий, видимых в Инспекторе 
объектов. Class Name – имя класса нового компонента. Pallete Page – стра-
ница Палитры компонентов для размещения нового компонента. Можно ука-
зать как существующую, так и новую страницу, которая будет автоматически 
создана. Unit File Name – имя файла с программным кодом компонента. 

Модуль-заготовка компонента, создаваемый IDE, содержит каркас описа-
ния класса компонента и процедуру RegisterComponents для регистрации 
компонента в IDE. Наша задача – наполнить этот модуль содержимым. 

Наш компонент должен содержать новое свойство, ответственное за цвет 
надписи. Назовем это свойство LabelColor. Его тип, очевидно, TColor. Так как 
оно должно быть видимо в Инспекторе объектов, объявим его в секции 

published. Для установки и чтения свойства используем методы 
SetLabelColor и GetLabelColor, описанные в секции private: 

unit NewLabel; 

interface 

uses Windows, Messages, SysUtils, Classes, Graphics, Controls, 

Forms, Dialogs, StdCtrls; 

type 

  TNewLabel = class(TLabel) 

  private 

    procedure SetLabelColor(AValue: TColor); 

    function GetLabelColor: TColor; 

  published 

    property LabelColor: TСolor 

                         read GetLabelColor write SetLabelColor; 

  end; 

procedure Register; 

implementation 

procedure TNewLabel.SetLabelColor; 

begin 

  if AValue <> LabelColor then Font.Color := AValue; 

end; 

procedure TNewLabel.GetLabelColor; 

begin 

  Result := Font.Color; 

end; 

procedure Register; 

begin 

  RegisterComponents('Samples', [TNewLabel]); 

end; 

end. 

Обратите внимание на синтаксис реализации методов и на разнесение ме-
тодов по разделам видимости. 

Разобраться с атрибутами классов компонентов, а также с разделами их 
видимости помогает Браузер объектов. Вызвать его можно после компиляции 
проекта, используя команду View|Browser. Браузер показывает дерево классов 
в левой части, правая часть содержит сгруппированные по разделам видимости 
атрибуты текущего класса. Для изучения компонентов используется также ис-
ходный код библиотеки VCL, поставляемый с редакцией Delphi Enterprise. 

14.2. Пакеты 
 

Процесс инсталляции компонента связан с понятием пакета. Пакет 
(package) – это расширенный вариант DLL, который может использоваться как 
программой, так и средой разработки Delphi. Фактически пакет – это несколько 
скомпилированных модулей (*.DCU) и, возможно, некоторые ресурсы, специ-
альным образом объединенные. Файлы пакетов имеют расширение BPL. 

Исходный файл пакета имеет расширение DPK и является обычным тексто-
вым файлом. Рассмотрим его структуру на примере: 

package Sample; 

{$R ‘COMP.DCR’} //подключение ресурсов (значок компонента) 

{$DESCRIPTION ‘Sample Components’} //описание пакета 

requires 

  vcl50; 

contains 

  Comp in ‘Comp.pas’; 

end. 

Директива package обозначает начало исходного файла пакета, директива 
requires – начало списка пакетов, требуемых для компиляции исходного. Ди-
ректива contains обозначает начало списка имен модулей, из которых состоит 
пакет. Имена модулей разделяются запятыми, каждый модуль представлен ли-
бо именем, либо именем модуля и именем файла в виде строки. 

Как правило, редактирование исходного текста пакета вручную не произ-
водится. Для этого применяется Менеджер пакетов из IDE Delphi. 

При помощи директивы компилятора {$DesignOnly On}, помещенной в 
исходный файл пакета, он может быть помечен как пакет разработки (design 
time packages). Такой пакет нельзя присоединить к приложению, а можно ис-
пользовать только при разработке приложения в IDE. По умолчанию директива 
выключена. При помощи директивы компилятора {$RunOnly On} пакет поме-
чается как пакет времени выполнения (runtime packages). Такой пакет нельзя за-
грузить в IDE. Директивы не являются взаимоисключающими. По умолчанию 
установлено {$DesignOnly Off} и {$RunOnly Off} (пакет можно использо-
вать и при разработке приложения и при выполнении). 

Управление пакетами можно осуществить, вызвав диалоговую панель 
Packages (Project|Options|Packages или Component|Install Packages). 
Панель содержит две части. В первой находится список используемых IDE па-
кетов дизайна. Мы можем временно отключить использование пакета, а значит, 
и входящих в него компонент, убрать пакет из IDE (кнопка Remove) или доба-
вить новый скомпилированный пакет с компонентами (кнопка Add). Состав 
компонентов, входящих в пакет, можно изучить при помощи кнопки 
Components. 

Вторая часть диалоговой панели управляет генерацией кода с подключе-
нием пакетов времени выполнения. Установка флага Build with runtime 

packages позволяет сгенерировать приложение, которое не включает в себя код 
используемых компонентов и классов, а берет его во время выполнения из па-
кетов, перечисленных в диалоговом окне строкой ниже. Такое приложение бу-
дет малым по размеру (простейшее – около 12 Кбайт), но для своей работы оно 
будет требовать указанные в строке диалога пакеты. Главный пакет, содержа-
щий основные компоненты, называется в Delphi 3 VCL30.BPL и имеет размер 
около 3,5 Мбайт. 

Вернемся к нашему компоненту и завершим его установку. Для этого тре-
буются создание DCU–файла для модуля компонента, включение этого файла в 
существующий пакет или создание нового пакета, компиляция и инсталляция 
полученного пакета. Совокупность этих действий выполняется при вызове ко-
манды Component|Install Component. После выполнения команды в появив-
шемся диалоговом окне можно выбрать инсталляцию в новый или существую-
щий пакет. Выберем инсталляцию в новый пакет (Into new package), укажем 
имя модуля (Unit file name), имя пакета (Package file name, должно отли-
чаться от имени модуля), описание пакета (Package description, заполнять 
не обязательно). После закрытия окна произойдут немедленная компиляция и 
установка пакета. На Палитре компонентов появится новый компонент. 

Более сложный путь предполагает создание пакета вручную. Для этого 
выполняется команда File|New|Package. Появившееся окно содержит кнопки 
для добавления или удаления модулей в пакет, компиляции пакета, установки 
пакета и компонент на палитру. 

Если компонент активно редактируется в процессе работы, желательно 
тестировать его, не прибегая к инсталляции. Пусть компонент хранится в файле 
c:\work\NewLabel.pas. Создадим маленький тестовый проект, содержащий 
форму и кнопку, при нажатии на которую на форме появляется компонент. Для 
этого достаточно в секции uses модуля формы подключить NewLabel.pas, а в 
обработчике нажатия кнопки создать и отобразить компонент: 

procedure TForm1.Button1Click(Sender: TObject); 

var L: TNewLabel; 

begin 

  L := TNewLabel.Create(Self); //создаем компонент 

  L.Caption := 'Hi–Hi'; //настраиваем его 

  L.LabelColor := clRed; 

  L.Parent := Form1; //теперь отображаем 

end; 
 

14. 3. Компонент TClock 
 

Рассмотрим еще один пример. Создадим компонент «Цифровые часы» 
(TClock) на основе текстовой метки. Для того чтобы не перегружать Инспектор 

объектов свойствами, выберем в качестве класса-предка TCustomLabel. Изна-
чально у такого компонента вообще не будет видимых в Инспекторе объектов 
событий и очень мало свойств. Это происходит потому, что полезные свойства 
и события в классе TCustomLabel скрыты в секции protected, их нужно пере-
нести в секцию published. Работа компонента TClock будет опираться на ком-
понент TTimer. Основные свойства этого компонента: Enabled – включение 
таймера, Interval – через сколько миллисекунд происходит событие таймера 
OnTimer. 

Наш компонент будет иметь поле для хранения объекта класса TTimer; 
создавать этот объект, настраивать его свойства и событие мы будем в конст-
рукторе TClock.Create. Кроме этого, к модулю компонента необходимо под-
ключить модуль extctrls, содержащий класс TTimer. Компонент TClock бу-
дет обладать собственным событием OnChange, которое происходит каждую 
секунду при включенном таймере: 

unit Clock; 

interface 

uses Windows, Messages, SysUtils, Classes, Graphics, Controls, 

Forms, Dialogs, StdCtrls, extctrls; 

type 

  TClock = class(TCustomLabel) 

  private 

    fTimer: TTimer; //содержит объект–таймер 

    fOnChange: TNotifyEvent; //хранит обработчик события 

    procedure TimerTick(Sender: TObject); //обновляет показания 

    procedure SetEnabled(AValue: Boolean); 

    function GetEnabled: Boolean; 

  public 

    constructor Create(AOwner: TComponent);override; 

  published 

    property Enabled: Boolean 

                  read GetEnabled write SetEnabled default True; 

    property Font; //перенесено из секции protected 

    property OnMouseMove; //перенесено из секции protected 

    property OnChange: TNotifyEvent 

                       read FOnChange write FOnChange; 

  end; 

. . .  

constructor TClock.Create; 

begin 

  inherited; 

  fTimer := TTimer.Create(Self); //создаем таймер 

  fTimer.Interval := 1000; //настраиваем его 

  fTimer.OnTimer := TimerTick; 

  fTimer.Enabled := True; 

  TimerTick(Self); //обновим показания часов 

end; 

procedure TClock.TimerTick; 

begin 

  Caption := TimeToStr(Time); //выведем текущее время 

  if Assigned(fOnChange) then fOnChange(Self); 

end; 

procedure TClock.SetEnabled; 

begin 

  if AValue <> fTimer.Enabled then 

  begin 

    fTimer.Enabled := AValue; //добираемся до таймера 

    if fTimer.Enabled then TimerTick(Self); //обновим показания 

  end 

end; 

procedure TClock.GetEnabled; 

begin 

  Result := fTimer.Enabled 

end; 

end; 

Этот пример продемонстрировал наследование от классов вида 
TCustomXXX, перенос свойств из одной области видимости в другую и создание 
собственных обработчиков событий. 

 
14. 4. Компоненты, наследники TGraphicControl 

 
Следующий пример – рисуемый компонент, наследник класса Tgraphic-

Control. Напоминаем, что класс TGraphicControl имеет виртуальный метод 
Paint. Его можно перекрывать для собственного рисования компонента. В ме-
тоде Paint доступно свойство класса Canvas. Опишем компонент для вывода 
горизонтальной стрелки: 

unit Arrow; 

. . . 

type 

  TDirection = (drLeft, drRight); 

  TArrow = class(TGraphicControl) 

  private 

    fDirection: TDirection; 

    procedure SetDirection(ADirection: TDirection); 

  protected 

    procedure Paint;override; 

  public 

    constructor Create(AOwner: TComponent);override; 

  published 

    property Height default 20; 

    property Width default 60; 

    property Color; 

    property Direction: TDirection 

             read fDirection write SetDirection default drRight; 

  end; 

constructor TArrow.Create; 

begin 

  inherited; 

  Height := 20; 

  Width := 60; 

  Direction := drRight 

end; 

procedure TArrow.SetDirection; 

begin 

  if ADirection <> fDirection then 

  begin 

    fDirection := ADirection; 

    Invalidate 

  end 

end; 

procedure TArrow.Paint; 

var dx, dy: Integer; 

begin 

  inherited; 

  dx := Width div 6; 

  dy := Height div 2; 

  with Canvas do 

  begin 

    if csDesigning in ComponentState then 

    begin 

      Pen.Style := psDash; 

      Brush.Style := bsClear; 

      Rectangle(0, 0, Width, Height) 

    end; 

    Pen.Style := psSolid; 

    Pen.Color := Color; 

    case Direction of 

      drRight: PolyLine([Point(Width–dx,0), Point(Width,dy), 

Point(Width–dx,Height), Point(Width,dy), Point(0,dy)]); 

      drLeft: PolyLine([Point(dx,0), Point(0,dy), 

Point(dx,Height), Point(0,dy), Point(Width,dy)]); 

    end 

  end 

end; 

Обратим внимание на следующие моменты. Для задания направления 
стрелки введен перечисляемый тип TDirection. Работа с таким типом полно-
стью поддерживается Инспектором объектов. Свойства Width и Height перене-
сены из секции protected в секцию published с указанием нового значения по 
умолчанию. В методе Paint использовано свойство ComponentState класса 
TComponent, которое позволяет контролировать, в каком режиме находится 
компонент. В данном случае ComponentState используется для рисования 
пунктирной линии на границе компонента в режиме проектирования формы. 

 
14.5. Компоненты с обработкой сообщений 

 
Последний пример компонента демонстрирует использование методов об-

работки сообщений. В примере обрабатываются внутренние сообщения Delphi 
о проходе указателя мыши над компонентом. Реализована «активная кнопка», 
заголовок которой становится полужирным, когда на кнопку указывает мышь: 

unit ActiveButton; 

type 

  TActiveButton = class(TButton) 

  protected 

  procedure MouseEnter(var Msg: TMessage);message cm_mouseEnter; 

  procedure MouseLeave(var Msg: TMessage);message cm_mouseLeave; 

  end; 

procedure TActiveButton.MouseEnter(var Msg: TMessage); 

begin 

  Font.Style := Font.Style + [fsBold] 

end; 

procedure TActiveButton.MouseLeave(var Msg: TMessage); 

begin 

  Font.Style := Font.Style – [fsBold] 

end; 

15. Взаимодействие приложений. Работа с сообщениями 
 

В этом разделе будут описаны несколько приемов совместной работы при-
ложений, созданных в Delphi, и внешних программ. Приведенная информация 
не претендует на полноту и для более глубокого изучения необходима допол-
нительная литература. 

 
15.1. Запуск внешних программ 

 
Для запуска внешних программ из приложения, созданного в Delphi, мож-

но использовать следующую функцию: 
function WinExec(CmdLine: PChar; CmdShow: Integer): Integer; 

Параметры функции означают следующее: 
CmdLine – имя исполняемого файла с параметрами. Если это имя задано 

без пути, то файл ищется по правилам, определенным для поиска DLL; 
CmdShow – вид окна запускаемого приложения (одна из констант 

SW_RESTORE, SW_MAXIMIZE и т. п.). 
Если запуск программмы успешен, функция возвращает значение большее, 

чем 31. Меньшее значение говорит об ошибке (0 – мало памяти, 2 – файл не 
найден и т. п.). Примеры использования функции: 

WinExec(‘nc’, SW_RESTORE); // запуск Norton Commander 

WinExec(‘command.com’, SW_RESTORE); // запуск сеанса DOS 
Альтернативной возможностью является использование функции Shel-

lExecute(Wnd: HWnd; Operation, FileName, Parameters, Directory: 

PChar; CmdShow: Integer): Integer. По сравнению с WinExec она обладает 
более широкими возможностями, к примеру, может не только запустить при-
ложение, но и напечатать документ. Смысл параметров функции следующий: 

Wnd – дескриптор родительского окна (обычно указывают свойство 
Handle); 

Operation – операция (’open’ – открыть файл, ’print’ – распечатать 
файл, ’explore’ – открыть папку в Проводнике Windows); 

FileName - имя файла или папки; 
Parameters – параметры для файла (или nil); 
Directory – каталог по умолчанию; 
CmdShow – аналогично WinExec. 
Для возвращаемого ShellExecute значения справедливо то же, что и для 

WinExec. Приведем примеры использования функции: 
ShellExecute(Handle, ’print’, ’file.doc’, nil, nil, SW_RESTORE) 

//запуск редактора Word и печать файла file.doc 
ShellExecute(Handle, ’explore’, ’c:\Work’, nil, nil, SW_RESTORE) 

//откроет папку c:\Work в Проводнике 

Функция FindExecutable(FileName,Directory,Buffer:PChar): THan-
dle позволяет получить имя exe-файла, связанного с программой, указанной в 
параметре FileName, и заносит ответ в параметр Buffer: 

var Answ: array[0..254] of Char; //переменная для записи ответа 

. . . 

FindExecutable(‘file.doc’,nil,Answ); //Answ =’C:\…\winword.exe’ 
 

15.2. Базовые моменты технологии COM 
 

Под технологий COM (Component Object Mode) понимается объектно-
ориентированный механизм интеграции приложений. По сути, эта технология 
позволяет одним приложениям предоставить для работы свои внутренние объ-
екты, а другим приложениям управлять этими объектами. COM-сервер – это 
приложение, объекты которого доступны другим приложениям (все такие при-
ложения зарегистрированы в реестре, так что их можно найти по уникальным 
именам). COM-контроллер (или COM-клиент) – приложение, которое обраща-
ется к объектам COM-сервера. 

Рассмотрим простейший пример кода COM-клиента, который работает с 
COM-сервером Microsoft Word: 

procedure TForm1.Button1Click(Sender: TObject); 

var MSWord: Variant; //переменная для доступа к COM-серверу 

begin 

  MSWord := CreateOLEObject('Word.Basic'); 

  //создание объекта для доступа к COM-серверу 

  MSWord.FileNewDefault; //создаем новый документ 

  MSWord.Insert('Hello from Delphi'); //вставляем в него текст 

  MSWord.FileSaveAs('C:\Work\hello.doc'); //сохраняем в файле 

  MSWord := UnAssigned; //освобождаем объект 

end; 

Функция CreateOLEObject объявлена в модуле COMObj. Методы, которые 
мы вызываем у MSWord, являются не методами Object Pascal, а методами VBA, 
они расположены в другом приложении и могут иметь непаскалевский синтак-
сис. Приведенный пример несколько устарел. Начиная с Delphi 3, для подобных 
целей используются так называемые интерфейсные типы (interface types). 

 
15.3. Сообщения. Обработка и перехват сообщений 

 
Дадим некоторые определения: 
Дескриптор (Handle) – уникальный целочисленный идентификатор, позво-

ляющий системе однозначно определить некий объект (окно, файл, ресурс); 
Сообщение (Message) – системная структура данных, содержащая несколь-

ко числовых полей. Служит для взаимодействия системы и приложений, а так-

же приложений между собой и с отдельными частями одного приложения. Сис-
темные сообщения Windows однозначно определяются своим номером (одно из 
полей структуры). 

В Delphi дескрипторам соответствуют собственные типы THandle, HDC, 
HWND (фактически по формату совпадают с типом Integer), а сообщениям сис-
темы – тип TMsg. 

Сделаем небольшое отступление и рассмотрим основные этапы создания 
программы для Windows без VCL. Такую программу можно написать в Delphi, 
этот процесс напоминает создание консольного приложения, для которого по-
надобятся два модуля – windows и messages. Первый является интерфейсным 
модулем к набору DLL, содержащих системные функции Windows, а второй 
содержит именованные константы для различных сообщений (чтобы не упот-
реблять их числовые идентификаторы) и несколько полезных типов данных. 
Подробно программа воспроизводиться не будет, так как ее текст достаточно 
большой (40–60 строк). Структура программы совпадает со структурой любого 
Windows-приложения: 

1) регистрация класса окна; 
2) создание окна; 
3) цикл обработки сообщений (взять сообщение из очереди – оттранслиро-

вать – направить в систему, продолжать цикл до получения сообщения о выхо-
де из программы); 

4) оконная функция, в которую, в конце концов, попадают все сообщения 
и которая содержит оператор case для распознавания конкретных сообщений. 

Все моменты здесь ключевые, подробнее они изучаются при прохождении 
курса «Системное программирование». 

Попробуем разобраться, в каких местах Delphi скрыты эти ключевые мо-
менты. В простейшей программе, созданной в Delphi, как правило, существует 
несколько элементов, соответствующих окнам в терминах Windows (кнопка, 
окно формы, список и т.д.). Все такие элементы являются потомками класса 
TWinControl и теоретически могут получать сообщения от системы. При этом 
сообщение системы проходит по следующей цепочке: 

1) вызывается функция окна MainWndProc; 
2) MainWndProc вызывает метод, адрес которого находится в свойстве 

WindowProc. По умолчанию это свойство содержит адрес виртуальной проце-
дуры WndProc; 

3) WndProc реализует обработку некоторых сообщений, после такой обра-
ботки вызывается метод Dispatch; 

4) Dispatch ищет динамический метод-обработчик сообщения по иденти-
фикатору сообщения; 

5) если Dispatch не нашел обработчик, он вызывает метод Default-
Handler, который передает сообщения обработчику сообщений по умолчанию 
из Windows. 

Все вышеуказанные методы работают с сообщениями не типа TMsg, а типа 
TMessage. Кроме объявления TMessage в модуле messages содержатся объявле-

ния типов, идентичных TMessage по длине, но интерпретирующих данные типа в 
виде различных структур для упрощения доступа к параметрам сообщения. 

Сам цикл обработки сообщений инкапсулирован в методе TApplica-
tion.ProcessMessage. В ходе работы этого метода вызывается обработчик со-
бытия TApplication.OnMessage 

Теперь опишем возможности перехвата и реакции на сообщения: 
1. Перекрытие виртуального метода WndProc окна (формы, кнопки). При 

этом не забывать вызвать перекрытый обработчик. 
2. Написание своего обработчика сообщений (с ключевым словом 

message). 
3. Перекрытие виртуального метода DefaultHandler (редкий подход) 
4. Написание обработчика события OnMessage для перехвата сообщений, 

поставленных в очередь приложения. 
Рассмотрим следующий пример. Будем отлавливать щелчки правой кноп-

ки мыши на форме. В проекте используются вышеописанные возможности пе-
рекрытия: 

type 

  TForm1 = class(TForm) 

    Label1: TLabel; 

    procedure FormMouseDown(Sender: TObject; Button: TMouseBut-

ton; Shift: TShiftState; X, Y: Integer); 

    procedure FormCreate(Sender: TObject); 

  public 

    procedure WndProc(var Message: TMessage); override; 

    procedure WmLButDown(var Message: TWMMouse);message 

WM_LBUTTONDOWN; 

    procedure DefaultHandler(var Message); override; 

    procedure AppMessage(var Msg: TMsg; var Handled: Boolean); 

  end; 

. . . 

procedure TForm1.WndProc(var Message: TMessage); 

begin 

  if Message.Msg = wm_LButtonDown then ShowMessage('WndProc'); 

  inherited 

end; 

procedure TForm1.WmLButDown(var Message:TWMMouse); 

begin 

  ShowMessage('WmLButDown'); 

  inherited 

end; 

procedure TForm1.DefaultHandler(var Message); 

begin 

  if TMessage(Message).Msg = wm_LButtonDown then ShowMes-

sage('DefHandler'); 

  inherited; 

end; 

procedure TForm1.FormMouseDown(Sender: TObject; Button: TMouse-

Button; Shift: TShiftState; X, Y: Integer); 

begin 

  if  Button = mbLeft then ShowMessage('OnMouseDown') 

end; 

procedure TForm1.AppMessage(var Msg: TMsg; var Handled: Boo-

lean); 

begin 

  if Msg.message = WM_LBUTTONDOWN then begin 

    Label1.Caption := 'OnMessage'; 

    Handled := False; 

  end; 

end; 

Обратите внимание на последовательность, с которой происходит обра-
ботка сообщений: OnMessage – WndProc – WmLButDown – DefaultHandler – On-
MouseDown. Так же обратите внимание на использование специальных типов и 
констант для распознавания сообщений. 

К перехватчикам сообщений (WmLButDown) предъявляются следующие 
требования: перехватчик сообщения всегда является процедурой, которая 
должна иметь единственный var-параметр; после имени процедуры указыва-
ются директива message и один числовой параметр (или константа), соответст-
вующий обрабатываемому сообщению; в перехватчике, как правило, вызывает-
ся унаследованный метод. 

Кроме системных сообщений компоненты Delphi определяют много собствен-
ных сообщений, которые также можно обработать перехватчиками сообщений. 

 
15.4. Посылка сообщений. Управление окнами 

 
Приведем несколько функций, при помощи которых можно работать с ок-

нами и компонентами, используя сообщения. Во всех функциях, как правило, 
одним из параметров выступает дескриптор окна. Если компонент порожден от 
класса TWinControl, то он имеем соответствующее свойство Handle. Иначе 
можно попытаться узнать дескриптор через вызов функции: 

function FindWindow(ClassName, WndName: PChar): HWnd 

Здесь ClassName – имя класса окна, WndName – его заголовок. Для форм и 
компонент Delphi эти параметры определяются просто (TForm1, Form1). Для 
других окон может помочь входящая в состав Delphi утилита WinSight32. 

Имея дескриптор, можно использовать, например, такие функции: 
function CloseWindow(hWnd: HWnd): BOOL – сворачивает окно; 
function EnableWindow(hWnd: HWnd; en: BOOL): BOOL – делает окно 

доступным или недоступным. 
Для посылки сообщений можно использовать функцию: 

function SendMessage(hWnd: HWnd; Msg, wParam: word; lParam: Lon-

gint): Longint, 

которая осуществляет посылку сообщения Msg с параметрами wParam и lParam 
окну с дескриптором hWnd. Функция ждет, пока сообщение не будет обработано. 

Функция 
function PostMessage(hWnd: HWnd; Msg,wParam: word; lParam: Lon-

gint): Longint 

по параметрам и действию аналогична SendMessage, но передает сообщение не 
окну, а ставит его в очередь сообщений приложения. 

Метод класса TControl 
function Perform(Msg: Cardinal; WParam, LParam: Longint): Lon-

gint; 

посылает сообщение тому компоненту, у которого вызывается (через вызов его 
оконной функции). 

